Auxiliary Variables in TLA$^+$

Leslie Lamport and Stephan Merz

27 May 2017
Abstract

Auxiliary variables are often needed for verifying that an implementation is correct with respect to a higher-level specification. They augment the formal description of the implementation without changing its semantics—that is, the set of behaviors that it describes. This paper explains rules for adding history, prophecy, and stuttering variables to TLA\(^+\) specifications, ensuring that the augmented specification is equivalent to the original one. The rules are explained with toy examples, and they are used to verify the correctness of a simplified version of a snapshot algorithm due to Afek et al.
1 Introduction

With state-based methods, checking that an implementation satisfies a higher-level specification requires describing how the higher-level concepts in the specification are represented by the lower-level data structures of the implementation. This approach was first proposed in the domain of sequential systems by Hoare in 1972 [5]. Hoare called the description an abstraction function. The generalization to concurrent systems was called a refinement mapping by Abadi and Lamport [2]. They observed that constructing a refinement mapping may require adding auxiliary variables to the implementation—variables that do not alter the behavior of the actual variables and need not be implemented.

This paper is about adding auxiliary variables to TLA\(^+\) specifications. The ideas we present should be applicable to other state-based specification methods, but we make no attempt to translate them into those other methods. We hope that a future paper will present the basic ideas in a language-independent way and will contain soundness and completeness proofs. Our goal here is to teach engineers writing TLA\(^+\) specifications how to add auxiliary variables when they need them.

We assume the reader can understand TLA\(^+\) specifications. A basic understanding of refinement mappings will be helpful but isn’t necessary. TLA\(^+\) and refinement mappings are explained in the book Specifying Systems [8] and in material listed on the TLA web page [7].

This is a long paper, in part because it contains 25 figures with actual TLA\(^+\) specifications. The paper contains hyperlinks, and we recommend reading the pdf version on line. If you are doing that, you can download the source files for all the TLA\(^+\) specifications described in this paper by clicking here. Otherwise, you can find the URL in the reference list [6]. We expect that engineers will have to study the specifications carefully to learn how to add auxiliary variables to their specifications.

We explain three kinds of auxiliary variables: history, prophecy, and stuttering variables. History variables record information about the system’s past behavior. They have been used since at least the 1970s [9]. They were sometimes called “ghost” variables. Prophecy variables predict the future behavior of the system. They were introduced by Abadi and Lamport in 1991 [2]. The need for them was also implicit in an example presented in Herlihy and Wing’s classic paper on linearizability [4]. We found the original prophecy variables very difficult to use in practice. The prophecy variables described here are new, and our experience with them so far indicates that they are reasonably easy to use in practice. Stuttering variables add “stuttering” steps—ones that leave the specification’s actual variables unchanged. Abadi and Lamport originally used prophecy variables to add stuttering steps, but we have found it better to introduce stuttering steps with a separate kind of variable.

We will mostly ignore liveness and consider only safety specifications. The
canonical form of a TLA$^+$ specification consists of a safety specification of the form $\text{Init} \land \Box[\text{Next}]_{\text{vars}}$ conjoined with a liveness condition. An auxiliary variable is added by modifying the safety specification, but leaving the liveness condition unchanged. Liveness therefore poses no problem for auxiliary variables and is discussed only briefly.

## 2 Refinement Mappings

We will illustrate refinement mappings with a simple, useless example. A user presents a server with a sequence of integer inputs. The server responds to each input value $i$ with one of the following outputs: $Hi$ if $i$ is the largest number input so far, $Lo$ if it’s the smallest number input so far, $Both$ if it’s both, and $None$ if it’s neither. We declare $Hi, Lo, Both, and None$ in a constants statement. They are assumed not to be integers.

### 2.1 Specification $MinMax1$

Our first specification appears in a module named $MinMax1$. It describes the interaction of the user and the server with two variables: a variable $x$ to hold an input or a response, and a variable $turn$ that indicates whether it’s the user’s turn to input a value or the server’s turn to respond. The specification also uses a variable $y$ to hold the set of values input so far. The initial predicate is

\[
\text{Init} \triangleq \land x = \text{None} \\
\land turn = \text{"input"} \\
\land y = \{\}
\]

The next-state relation $\text{Next}$ equals $\text{InputNum} \lor \text{Respond}$ where $\text{InputNum}$ is the user’s input action and $\text{Respond}$ is the server’s output action. The definition of $\text{InputNum}$ is simple:

\[
\text{InputNum} \triangleq \land turn = \text{"input"} \\
\land turn' = \text{"output"} \\
\land x' \in \text{Int} \\
\land y' = y
\]

To define the $\text{Respond}$ action, we must first define operators $\text{setMax}$ and $\text{setMin}$ so that, for any finite nonempty set $S$ of integers, $\text{setMax}(S)$ and $\text{setMin}(S)$ are the maximum and minimum element, respectively, of $S$. The definitions are:

\[
\text{setMax}(S) \triangleq \text{choose } t \in S : \forall s \in S : t \geq s \\
\text{setMin}(S) \triangleq \text{choose } t \in S : \forall s \in S : t \leq s
\]

The definition of $\text{Respond}$ is:
\[\text{Respond} \triangleq \begin{array}{c}
\wedge \text{turn} = \text{"output"} \\
\wedge \text{turn}' = \text{"input"} \\
\wedge y' = y \cup \{x\} \\
\wedge x' = \text{IF } x = \text{setMax}(y') \text{ THEN IF } x = \text{setMin}(y') \text{ THEN Both ELSE Hi} \\
\text{ELSE IF } x = \text{setMin}(y') \text{ THEN Lo ELSE None}
\end{array}\]

Note that action InputNum is enabled iff \text{turn} equals “input”, and action Respond is enabled iff \text{turn} equals “output”. The complete specification is the formula

\[\text{Spec} \triangleq \text{Init} \land \Box([\text{Next}]_{\text{vars}})\]

where \text{vars} is the tuple \langle x, \text{turn}, y \rangle of variables. The module MinMax1 we have written thus far is shown in Figure 1.

### 2.2 The Hiding Operator $\exists$

Recall that a behavior is a sequence of states, where a state is an assignment of values to all possible variables. For specification \text{Spec} of module MinMax1, the interesting part of the state is the assignment of values to \text{x}, \text{turn}, and \text{y}. Our specification allows all other variables to have any value at any state of any behavior.

The purpose of this specification is to describe the interaction of the user and the server. This interaction is described by the values of \text{x} and \text{turn}. The value of \text{y} is needed only to describe how the values of \text{x} and \text{turn} can change. We consider \text{x} and \text{turn} to be the externally visible or observable values of the specification and \text{y} to be an internal variable. A philosophically correct specification of our user/server system would allow only behaviors in which the values of \text{x} and \text{turn} are as specified by \text{Spec}, but would not constrain the value of \text{y}. We can write such a specification in terms of the temporal-logic operator $\exists$.

For any temporal formula $F$ and variable $v$, the formula $\exists v : F$ is defined approximately as follows. A behavior $\sigma$ satisfies $\exists v : F$ iff there exists a behavior $\tau$ satisfying $F$ such that $\tau$ is identical to $\sigma$ except for the values its states assign to $v$. The precise definition is more complicated because a temporal formula of TLA$^+$ may neither require nor prohibit stuttering steps, but we will use the approximate definition for now. The operator $\exists$ is much like the ordinary existential quantifier $\exists$ except that $\exists v : F$ asserts the existence not of a single value for $v$ that makes $F$ true but rather of a sequence of values, one for each state in the behavior, that makes $F$ true on the behavior. This temporal existential quantifier $\exists$ satisfies most of the properties of ordinary quantification. For example, if the variable $v$ does not occur in formula $F$, then $\exists v : F$ is equivalent to $F$. We sometimes read the formula $\exists v : F$ as “$F$ with $v$ hidden”.

The philosophically correct specification of the MinMax1 system should consist of formula $Spec$ with $y$ hidden. The obvious way to write this specification is $\exists y : Spec$. However, we can’t do that for the following reason. Suppose
**Module MinMax1**

EXTENDS `Integers`

\[\text{setMax}(S) \triangleq \text{choose } t \in S : \forall s \in S : t \geq s\]
\[\text{setMin}(S) \triangleq \text{choose } t \in S : \forall s \in S : t \leq s\]

CONSTANTS `Lo`, `Hi`, `Both`, `None`

ASSUME \{`Lo`, `Hi`, `Both`, `None`\} \cap `Int` = `{}`

VARIABLES `x`, `turn`, `y`

\[\text{vars} \triangleq (x, \text{turn}, y)\]

\[\text{Init} \triangleq \wedge x = \text{None} \]
\[\wedge \text{turn} = \text{“input”} \]
\[\wedge y = \{\}\]

\[\text{InputNum} \triangleq \wedge \text{turn} = \text{“input”} \]
\[\wedge \text{turn}' = \text{“output”} \]
\[\wedge x' \in \text{Int} \]
\[\wedge y' = y\]

\[\text{Respond} \triangleq \wedge \text{turn} = \text{“output”} \]
\[\wedge \text{turn}' = \text{“input”} \]
\[\wedge y' = y \cup \{x\} \]
\[\wedge x' = \text{if } x = \text{setMax}(y') \]
\[\text{THEN if } x = \text{setMin}(y') \text{ THEN Both ELSE Hi} \]
\[\text{ELSE if } x = \text{setMin}(y') \text{ THEN Lo ELSE None}\]

\[\text{Next} \triangleq \text{InputNum} \lor \text{Respond}\]

\[\text{Spec} \triangleq \text{Init} \land \Box[\text{Next}]\text{vars}\]

---

Figure 1: Module MinMax1.

a module \(M\) defines \(exp\) to equal some expression. TLA\(^+\) does not allow the expression

\[(2.1) \{v \in exp : v^2 > 42\}\]

to appear at any point in module \(M\) where \(v\) is already declared or defined. Since \(exp\) must be defined for the expression to have a meaning, this means that \((2.1)\) is illegal if \(v\) is a declared variable that appears in the definition of \(exp\). Similarly, the formula \(\exists y : \text{Spec}\) is illegal because \(y\) appears in the definition of \(\text{Spec}\).\(^1\)

\(^1\)There are languages for writing math precisely that allow expression \((2.1)\) even if \(v\) is
There are ways to write the formula \( Spec \) with \( v \) hidden in TLA\(^+\). The most convenient ones involve writing it in another module that instantiates module \( MinMax1 \). Chapter 4 of \( Specifying \) \( Systems \) \[8\] explains one way to do this. However, there’s little reason to do it since the TLA\(^+\) tools cannot check specifications written with \( \exists \). (The TLAPS proof system may eventually be able to reason about it.) Instead, we take the formula \( \exists y : Spec \) to be an abbreviation for the formula \( \exists y : [Spec] \), where \([Spec]\) is the formula obtained from \( Spec \) by expanding all definitions. Formula \([Spec]\) contains only: TLA\(^+\) primitives; the constants \( Hi, Lo, Both, \) and \( None \); and the variables \( x, turn, \) and \( y \). Thus \( \exists y : Spec \) is meaningful in a context in which \( x \) and \( turn \) are declared variables. If used in a context in which \( y \) already has a meaning, we interpret \( \exists y : Spec \) to be the formula obtained from \( \exists y : [Spec] \) by replacing \( y \) everywhere with a new symbol.

What it means to expand all definitions in an expression is not as simple as it might seem. Consider the following definition:

\[
(2.2) \quad \text{NotUnique}(a) \triangleq \exists i : i \neq a
\]

It’s clear that the following theorem is true:

\[
(2.3) \quad \text{THEOREM } \forall a : \text{NotUnique}(a)
\]

Now suppose we follow the definition of \( \text{NotUnique} \) with:

\[
(2.4) \quad \text{CONSTANT } i \quad \quad \text{THEOREM } \text{NotUnique}(i)
\]

Theorem (2.3) obviously implies the theorem of (2.4). However, a naive expansion of the definition of \( \text{NotUnique} \) tells us that \([\text{NotUnique}(i)]\) equals \( \exists i : i \neq i \), which equals \( \text{FALSE} \). The problem is clear: the bound identifier \( i \) in the definition of \( \text{NotUnique} \) is not the same \( i \) as the one declared in the \( \text{CONSTANT} \) declaration. The following definition of \( \text{NotUnique} \) is equivalent to (2.2)

\[
\text{NotUnique}(a) \triangleq \exists jku : jku \neq a
\]

and with the naive expansion of this definition, \([\text{NotUnique}(i)]\) equals the true formula \( \exists jku : jku \neq i \) of (2.4).

The easiest way to define the meaning of expanding all definitions in an expression is to consider (2.2) to define \( \text{NotUnique}(a) \) to equal something like \( \exists v_{\_743} : v_{\_743} \neq a \), where \( v_{\_743} \) is an identifier that cannot be used anywhere else. In general, every bound identifier in a definition is replaced by some unique identifier.

already declared. In such a language, \( \exists y : Spec \) would be equivalent to \( \exists w : Spec \) for any identifier \( w \), which means it would be equivalent to \( Spec \).
Recursive definitions are not a problem for complete expansion of definitions because in TLA$^+$, a recursive definition is just an abbreviation for a non-recursive one. For example

\[f[i \in \text{Nat}] \triangleq \text{if } i = 0 \text{ then } 1 \text{ else } i \ast f[i - 1]\]

is an abbreviation for

\[f \triangleq \text{choose } f : f = [i \in \text{Nat} \mapsto \text{if } i = 0 \text{ then } 1 \text{ else } i \ast f[i - 1]]\]

so the bound identifier $f$ to the right of the \(\triangleq\) is not the same symbol as the $f$ being defined. (A recursive operator definition is an abbreviation for a much more complicated ordinary definition.)

### 2.3 Specification \textit{MinMax2}

The specification of our system in module \textit{MinMax1} uses the variable $y$ to remember the set of all values that the user has input. Module \textit{MinMax2} specifies the same user/server interaction that remembers only the smallest and largest values input so far, using the variables $min$ and $max$. Representing the initial state, before any values have been input, is a little tricky. It would be simpler if the standard \textit{Integers} module defined a value $\infty$ such that $-\infty < i < \infty$ for all integers $i$. So, we will write the spec pretending that it did. Afterwards, we’ll describe how to obtain an actual TLA$^+$ spec.

The initial predicate of the specification is:

\[
\text{Init} \triangleq \land x = \text{None} \\
\land turn = \text{"input"} \\
\land min = \infty \\
\land max = -\infty
\]

The user’s \textit{InputNum} action is the same as for the \textit{MinMax1} specification, except it leaves $min$ and $max$ rather than $y$ unchanged:

\[
\text{InputNum} \triangleq \land turn = \text{"input"} \\
\land turn' = \text{"output"} \\
\land x' \in \text{Int} \\
\land \text{UNCHANGED } \langle min, max \rangle
\]

Here is the system’s \textit{Respond} action:

\[
\text{Respond} \triangleq \land turn = \text{"output"} \\
\land turn' = \text{"input"} \\
\land min' = \text{if } x \leq min \text{ then } x \text{ else } min \\
\land max' = \text{if } x \geq max \text{ then } x \text{ else } max \\
\land x' = \text{if } x = max' \\
\text{then if } x = min' \text{ then Both else Hi} \\
\text{else if } x = min' \text{ then Lo else None}
\]
As usual, the complete specification is
\[
Spec \triangleq \text{Init} \land \Box[\text{Next}]_{\text{vars}}
\]
where this time \( \text{vars} \) is the tuple \( \langle x, \text{turn}, \text{min}, \text{max} \rangle \) of variables.

To turn this into a TLA\(^+\) specification, we replace \( \infty \) and \( -\infty \) by two constants \( \text{Infinity} \) and \( \text{MinusInfinity} \). In the definition of \( \text{Respond} \), we replace \( x \leq \text{min} \) and \( x \geq \text{max} \) by \( \text{IsLeq}(x, \text{min}) \) and \( \text{IsGeq}(x, \text{max}) \), where \( \text{IsLeq} \) and \( \text{IsGeq} \) are defined by
\[
\text{IsLeq}(i, j) \triangleq (j = \text{Infinity}) \lor (i \leq j)
\]
\[
\text{IsGeq}(i, j) \triangleq (j = \text{MinusInfinity}) \lor (i \geq j)
\]
These definitions must be preceded by declarations or definitions of \( \text{Infinity} \) and \( \text{MinusInfinity} \). They can equal any values, except that they must not be equal and neither of them should be an integer—otherwise, the spec wouldn’t mean what we want it to mean. We could declare \( \text{Infinity} \) and \( \text{MinusInfinity} \) in a constant declaration and then add an \text{assume} statement asserting that they aren’t in \( \text{Int} \). However, we prefer to define them like this:
\[
\text{Infinity} \triangleq \text{choose } n : n \notin \text{Int}
\]
\[
\text{MinusInfinity} \triangleq \text{choose } n : n \notin (\text{Int} \cup \{\text{Infinity}\})
\]
This completes module \( \text{MinMax2} \), which is shown in its entirety in Figure 2.

As before, we consider \( x \) and \( \text{turn} \) to be the externally visible variables, and \( \text{min} \) and \( \text{max} \) to be internal variables. The philosophically correct specification, which hides the internal variables \( \text{min} \) and \( \text{max} \), is \( \exists \text{min, max} : \text{Spec} \). Of course, this is an abbreviation for \( \exists \text{min} : (\exists \text{max} : \text{Spec}) \), which is equivalent to \( \exists \text{max} : (\exists \text{min} : \text{Spec}) \).

### 2.4 The Relation Between the Two Specifications

Using the standard TLA\(^+\) naming convention, we have given the two specifications the same name \( \text{Spec} \). To distinguish them, let \( \text{Spec}_1 \) be the specification \( \text{Spec} \) of module \( \text{MinMax1} \) and \( \text{Spec}_2 \) be the \( \text{Spec} \) of module \( \text{MinMax2} \).

It should be clear that both specifications describe the same behavior of the external variables \( x \) and \( \text{turn} \). This means that if we hide the internal variable \( y \) of \( \text{Spec}_1 \) and the internal variables \( \text{min} \) and \( \text{max} \) of \( \text{Spec}_2 \), we should obtain equivalent specifications. More precisely, we expect to this to be true:

\[
(2.5) \quad (\exists y : \text{Spec}_1) \equiv (\exists \text{min, max} : \text{Spec}_2)
\]

This formula is equivalent to the conjunction of these two formulas

\[
(2.6) \quad (\exists y : \text{Spec}_1) \Rightarrow (\exists \text{min, max} : \text{Spec}_2)
\]
\[
(2.7) \quad (\exists \text{min, max} : \text{Spec}_2) \Rightarrow (\exists y : \text{Spec}_1)
\]
EXTENDS Integers, Sequences

CONSTANTS Lo, Hi, Both, None

ASSUME \{Lo, Hi, Both, None\} \cap \text{Int} = \{

Infinity \triangleq \text{CHOOSE } n : n \notin \text{Int}

MinusInfinity \triangleq \text{CHOOSE } n : n \notin (\text{Int} \cup \{\text{Infinity}\})

IsLeq(i, j) \triangleq (j = \text{Infinity}) \lor (i \leq j)

IsGeq(i, j) \triangleq (j = \text{MinusInfinity}) \lor (i \geq j)

VARIABLES $x, \text{turn, min, max}$

vars \triangleq \langle x, \text{turn, min, max} \rangle

Init \triangleq \& x = \text{None}
\& \text{turn} = \text{"input"}
\& \text{min} = \text{Infinity}
\& \text{max} = \text{MinusInfinity}

InputNum \triangleq \& \text{turn} = \text{"input"}
\& \text{turn}' = \text{"output"}
\& x' \in \text{Int}
\& \text{UNCHANGED} \langle \text{min, max} \rangle

Respond \triangleq \& \text{turn} = \text{"output"}
\& \text{turn}' = \text{"input"}
\& \text{min}' = \text{IF } IsLeq(x, \text{min}) \text{ THEN } x \text{ ELSE } \text{min}
\& \text{max}' = \text{IF } IsGeq(x, \text{max}) \text{ THEN } x \text{ ELSE } \text{max}
\& x' = \text{IF } x = \text{max}' \text{ THEN IF } x = \text{min}' \text{ THEN } \text{Both} \text{ ELSE } \text{Hi}
\text{ELSE IF } x = \text{min}' \text{ THEN } \text{Lo} \text{ ELSE } \text{None}

Next \triangleq \text{InputNum} \lor \text{Respond}

Spec \triangleq \text{Init} \& \Box[\text{Next}]_{\text{vars}}

Figure 2: Module MinMax2.

We verify (2.5) by separately verifying (2.6) and (2.7). We first consider (2.6).

Formula (2.6) asserts of a behavior $\sigma$ that if there exists some way of assigning values to $y$ in the states of $\sigma$ to make it satisfy $Spec_1$, then $\sigma$ satisfies $\exists \text{min, max} : Spec_2$. Since the variable $y$ does not appear in $\exists \text{min, max} : Spec_2$, changing the values of $y$ in the states of $\sigma$ doesn’t affect whether it satisfies that formula. This implies that to verify (2.6), it suffices to show that any behavior $\sigma$ that satisfies $Spec_1$ also satisfies $\exists \text{min, max} : Spec_2$. In other words, to verify
(2.6), it suffices to verify

\( (2.8) \quad \text{Spec}_1 \Rightarrow (\exists \min, \max : \text{Spec}_2) \)

To verify (2.8), we must show that for any behavior \( \sigma \) that satisfies \( \text{Spec}_1 \), there exists a way of assigning the resulting behavior \( \text{Spec}_2 \). A standard way of doing that is to find explicit expressions \( \overline{\min} \) and \( \overline{\max} \) such that, if in each state of a behavior we assign to the variables \( \min \) and \( \max \) the values of \( \overline{\min} \) and \( \overline{\max} \) in that state, then the resulting behavior satisfies \( \text{Spec}_2 \). We do this by showing that any behavior satisfying \( \text{Spec}_1 \) satisfies the formula obtained by substituting \( \overline{\min} \) for \( \min \) and \( \overline{\max} \) for \( \max \) in \( \text{Spec}_2 \). Let's write that formula \( \overline{\text{Spec}_2} \), emphasizing that we must expand all definitions in \( \text{Spec}_2 \) before substituting \( \overline{\min} \) for \( \min \) and \( \overline{\max} \) for \( \max \). So, we verify (2.8) by verifying

\( (2.9) \quad \text{Spec}_1 \Rightarrow \overline{\text{Spec}_2} \)

We can write formula \( \overline{\text{Spec}_2} \) (or more precisely, a formula equivalent to it) in module \texttt{MinMax1} as follows. We first add the statement

\[ M \triangleq \text{INSTANCE} \texttt{MinMax2} \text{ WITH } \min \leftarrow \overline{\min}, \max \leftarrow \overline{\max} \]

For every defined symbol \( \text{def} \) in module \texttt{MinMax2}, this statement defines \( M!\text{def} \) to be equivalent to \( \overline{\text{def}} \), the formula whose definition is obtained by substituting \( \overline{\min} \) for \( \min \) and \( \overline{\max} \) for \( \max \) in the formula obtained by expanding all definitions in the definition of \( \text{def} \) in \texttt{MinMax2}.\(^2\) This instance statement therefore defines \( M!\text{Spec} \) to be equivalent to \( \overline{\text{Spec}_2} \), allowing us to write (2.9) in module \texttt{MinMax1} as the theorem:

**THEOREM** \( \text{Spec} \Rightarrow M!\text{Spec} \)

We can write a TLA\(^+\) proof of this theorem and check it with the TLAPS theorem prover. We can also have TLC check this theorem by creating a model for module \texttt{MinMax1} with specification \( \text{Spec} \) that substitutes a finite set of integers for \( \text{Int} \) and checks the property \( M!\text{Spec} \). But before we can do that, we have to determine what the expressions \( \overline{\min} \) and \( \overline{\max} \) are in the instance statement.

Formula (2.9) asserts that in a behavior \( \sigma \) satisfying \( \text{Spec}_1 \), if in each state of \( \sigma \) we assign to \( \min \) and \( \max \) the values of \( \overline{\min} \) and \( \overline{\max} \) in that state, then the resulting behavior satisfies \( \text{Spec}_2 \). One way of thinking about this is that in a behavior satisfying \( \text{Spec}_1 \), the values of \( \overline{\min} \) and \( \overline{\max} \) simulate the values that \( \text{Spec}_2 \) requires \( \min \) and \( \max \) to assume.

A little thought reveals that \( \overline{\min} \) and \( \overline{\max} \) should be defined as indicated in this statement:

\(^2\)Note that the declared constants \( Hi, Lo, Both, \) and \( None \) of module \texttt{MinMax2} have been implicitly instantiated by the constants of the same name declared in \texttt{MinMax1}.
\[
\begin{aligned}
\text{Infinity} & \triangleq \text{CHOOSE } n : n \notin \text{Int} \\
\text{MinusInfinity} & \triangleq \text{CHOOSE } n : n \notin (\text{Int} \cup \{\text{Infinity}\})
\end{aligned}
\]

\[
M \triangleq \text{INSTANCE MinMax2}
\]

\[
\begin{aligned}
\text{WITH } min & \leftarrow \text{IF } y = \{\} \text{ THEN Infinity ELSE } \text{setMin}(y), \\
max & \leftarrow \text{IF } y = \{\} \text{ THEN MinusInfinity ELSE } \text{setMax}(y)
\end{aligned}
\]

Figure 3: Additions to module \textit{MinMax1}.

\[
\begin{aligned}
M \triangleq \text{INSTANCE MinMax2}
\end{aligned}
\]

\[
\begin{aligned}
\text{WITH } min & \leftarrow \text{IF } y = \{\} \text{ THEN Infinity ELSE } \text{setMin}(y), \\
max & \leftarrow \text{IF } y = \{\} \text{ THEN MinusInfinity ELSE } \text{setMax}(y)
\end{aligned}
\]

Of course, we need to define \textit{Infinity} and \textit{MinusInfinity} before we can write that statement. They should be defined to be the same values as in \textit{MinMax2}, so we just copy the definitions from that module into module \textit{MinMax1}. We have added the statements in Figure 3 to the bottom of the module in Figure 1.

### 2.5 Refinement In General

In general, we have two specs: \textit{Spec1} with variables \(x_1, \ldots, x_m, y_1, \ldots, y_n\), and \textit{Spec2} with variables \(x_1, \ldots, x_m, z_1, \ldots, z_p\). For compactness let \(x\) denote \(x_1, \ldots, x_m\), let \(y\) denote \(y_1, \ldots, y_n\) and let \(z\) denote \(z_1, \ldots, z_p\). We consider \(x\) to be the externally visible variables of both specifications, and we consider \(y\) and \(z\) to be internal variables.

The specifications with their internal variables hidden are written \(\exists y : \textit{Spec1}\) and \(\exists z : \textit{Spec2}\). To verify that \(\exists y : \textit{Spec1}\) implements \(\exists z : \textit{Spec2}\), we must show that for each behavior satisfying \textit{Spec1}, there is some way to assign values of the variables \(z\) in each state so that the resulting behavior satisfies \textit{Spec2}. We do that by explicitly specifying those values of \(z\) in terms of the values of \(x\) and \(y\). More precisely, for each \(z_i\) we define an expression \(\overline{z_i}\) in terms of the values of \(x\) and \(y\) and show that \textit{Spec1} implements \([\textit{Spec2}]\), the specification obtained by expanding all definitions in \textit{Spec2} and substituting \(z_1 \leftarrow \overline{z_1}, \ldots, z_p \leftarrow \overline{z_p}\) in the resulting formula. This substitution is called a refinement mapping; and if \textit{Spec1} implements \([\textit{Spec2}]\), then we say that \textit{Spec1} implements \textit{Spec2} under the refinement mapping.

The assertion that \textit{Spec1} implements \textit{Spec2} under the refinement mapping \(z_1 \leftarrow \overline{z_1}, \ldots, z_p \leftarrow \overline{z_p}\) can be expressed in TLA\(^+\) as follows. Suppose \textit{Spec1} is formula \textit{Spec1} in a module named \textit{Mod1}, and \textit{Spec2} is formula \textit{Spec2} in a module named \textit{Mod2}. (\textit{Spec1} and \textit{Spec2} can be the same identifier.) For some identifier
Id, we add the following statement to Mod1:\(^3\)

\[ \text{Id} \triangleq \text{INSTANCE Mod2 with } z_1 \leftarrow \overline{z_1}, \ldots, z_p \leftarrow \overline{z_p} \]

The assertion that Spec\(_1\) implements Spec\(_2\) under the refinement mapping can then be expressed in module Mod\(_1\) by the following theorem:

\[ \text{THEOREM } \text{Spec}1 \Rightarrow \text{Id!Spec}2 \]

This theorem asserts that in any behavior satisfying Spec\(_1\), the values of the expressions \(\overline{z_i}\) are values that Spec\(_2\) permits the variables \(z_i\) to have. The shape of the theorem makes it explicit that in TLA\(^+\), implementation is implication. The correctness of the theorem can be checked (but seldom completely verified) with TLC for module Mod\(_1\) having Spec\(_1\) as the specification and Id!Spec2 as the temporal property to be checked.

As we will see, it is sometimes the case that \(\exists y : \text{Spec}1\) implements \(\exists z : \text{Spec}2\) but there does not exist a refinement mapping under which Spec\(_1\) implements Spec\(_2\). In that case, it is almost always possible to construct the necessary refinement mapping by adding auxiliary variables to Spec\(_1\). Adding auxiliary variables \(a\) to the specification Spec\(_1\) means finding a specification Spec\(_1^a\) such that \(\exists a : \text{Spec}1^a\) is equivalent to Spec\(_1\). Showing that \(\exists y, a : \text{Spec}1^a\) implements \(\exists z : \text{Spec}2\) shows that \(\exists y : \text{Spec}1\) implements \(\exists z : \text{Spec}2\), since \(\exists y, a : \text{Spec}1^a\) equals \(\exists y : \exists a : \text{Spec}1^a\) which is equivalent to \(\exists y : \text{Spec}1\). Even though we can’t define the expressions \(\overline{z_i}\) in terms of \(x\) and \(y\), we may be able to define them in terms of \(x\), \(y\), and \(a\).

We will define three kinds of auxiliary variables: history variables that remember what happened in the past, prophecy variables that predict what will happen in the future, and stuttering variables that add stuttering steps (ones that don’t change \(x\) and \(y\)).

### 3 History Variables

#### 3.1 Equivalence of MinMax1 and MinMax2

Let us return to the notation of Section 2.4, so Spec\(_1\) is specification Spec of module MinMax1 and Spec\(_2\) is specification Spec of module MinMax2. We observed that \(\exists y : \text{Spec}1\) and \(\exists \text{min, max} : \text{Spec}2\) are equivalent, meaning that each implements (implies) the other. We found a refinement mapping under which Spec\(_1\) implements Spec\(_2\). To prove the converse implication, we want to find a refinement mapping under which Spec\(_2\) implements Spec\(_1\). This means defining an expression \(\overline{y}\) in terms of the variables \(x\), turn, min, and max such

\(^3\)If Mod\(_2\) has declared \text{CONSTANTS}, then the statement must also specify expressions to be substituted for those constants. A substitution of the form \(id \leftarrow id\) for an identifier \(id\) can be omitted from the WITH clause.
that the values of $x$, $\text{turn}$, and $\overline{y}$ in any behavior allowed by $Spec_1$ are values of $x$, $\text{turn}$, and $y$ allowed by $Spec_2$.

In a behavior of $Spec_1$, the value of $y$ is the set of all values input by the user. However, in a behavior of $Spec_2$, the variables $\text{min}$ and $\text{max}$ record only the smallest and largest input values. There is no way to reconstruct the set of all values input from the variables of $MinMax2$. So, there is no refinement mapping under which $Spec_2$ implements $Spec_1$. To solve this problem, we write another spec $Spec^h_2$ that is the same as $Spec_2$, except that it also constrains the behavior of another variable $h$. More precisely, if we hide $h$ in $Spec^h_2$, then we get a specification that’s equivalent to $Spec_2$. Expressed mathematically, this means $\exists h : Spec^h_2$ is equivalent to $Spec_2$.

The initial predicate and next-state action of $Spec^h_2$ are the same as those of $Spec_2$, except they also describe the values that $h$ may assume. In particular, the value of $h$ records information about previous values of the variable $x$, but does not affect the current or future values of $x$ or any of the other variables $\text{turn}$, $\text{min}$, and $\text{max}$ of $Spec_2$. Thus $\exists h : Spec^h_2$ is equivalent to $Spec_2$. We call $h$ a history variable.

We write $Spec^h_2$ as follows in a TLA+ module $MinMax2H$. The module begins with the statement

```
EXTENDS MinMax2
```

that simply imports all the declarations and definitions from $MinMax$, defining $Spec$ to be the specification we are calling $Spec_2$. The module declares the variable $h$ and defines the initial predicate $InitH$ of $Spec^h_2$ by

$$InitH \triangleq Init \land (h = \{\})$$

The next-state action $NextH$ is defined to equal $InputNumH \lor RespondH$ where $InputNumH$ and $RespondH$ are defined as follows:

$$InputNumH \triangleq \land InputNum$$
$$\land h' = h$$

$$RespondH \triangleq \land Respond$$
$$\land h' = h \cup \{x\}$$

The specification $Spec^h_2$ is the following formula defined in the module:

$$SpecH \triangleq InitH \land \Box[NextH]_{varsH}$$

where $varsH$ equals $(vars, h)$. (Because $vars$ equals $(x, \text{turn}, \text{min}, \text{max})$, we can also define $varsH$ to equal $(x, \text{turn}, \text{min}, \text{max}, h)$; the two definitions give equivalent expressions UNCHANGED $varsH$.)

It’s easy to see that this specification asserts that $h$ is always equal to the set of all values that the user has input thus far, which is exactly what $Spec_1$ asserts about $y$. Therefore, $Spec^h_2$ implements $Spec_1$ under the refinement mapping 14
Module \textit{MinMax2H}

\begin{verbatim}
EXTENDS MinMax2

VARIABLE \( h \)
\( \text{varsH} \triangleq \langle \text{vars}, h \rangle \)
\( \text{InitH} \triangleq \text{Init} \land (h = \{\}) \)
\( \text{InputNumH} \triangleq \land \text{InputNum} \land h' = h \)
\( \text{RespondH} \triangleq \land \text{Respond} \land h' = h \cup \{x\} \)
\( \text{NextH} \triangleq \text{InputNumH} \lor \text{RespondH} \)
\( \text{SpecH} \triangleq \text{InitH} \land \Box[\text{NextH}]_{\text{varsH}} \)
\( M \triangleq \text{INSTANCE MinMax1 WITH } y \leftarrow h \)

THEOREM \( \text{SpecH} \Rightarrow M \! \text{Spec} \)
\end{verbatim}

\begin{figure}[h]
\centering
\includegraphics[width=\textwidth]{image}
\caption{Module \textit{MinMax2H}.}
\end{figure}

\( y \leftarrow h \)—that is, with \( \overline{y} \) equal to the expression \( h \). We express this in module \textit{MinMax2H} by

\( M \triangleq \text{INSTANCE MinMax1 WITH } y \leftarrow h \)

THEOREM \( \text{SpecH} \Rightarrow M \! \text{Spec} \)

The complete module \textit{MinMax2H} is in Figure 4

\subsection*{3.2 Disjunctive Representation}

The generalization from the \textit{MinMax} example is intuitively clear. We add a history variable \( h \) to a specification by conjoining \( h = \text{exp}_{\text{Init}} \) to its initial predicate and \( (h' = \text{exp}_A) \) to each subaction \( A \) of its next-state action, where expression \( \text{exp}_{\text{Init}} \) can contain the spec’s variables and each expression \( \text{exp}_A \) can contain the spec’s variables, both primed and unprimed, and \( h \) (unprimed). To make this precise, we have to state exactly what a \textit{subaction} is.

In general, there may be many different ways to define the subactions of a next-state action. In defining \textit{SpecH} in module \textit{MinMax2H}, we took \textit{InputNum} and \textit{Respond} to be the subactions of the next-state action \textit{Next} of \textit{MinMax2}. However, we can also consider \textit{Next} itself to be a subaction of \textit{Next}. We can do this and get an equivalent specification \textit{SpecH} by defining \textit{NextH} as follows:
The two specifications are equivalent because the two definitions of \( \text{NextH} \) are equivalent. Their equivalence is asserted by adding the following theorem to module \( \text{MinMax2H} \). The TLAPS proof system easily checks its by proof.

\[
\text{THEOREM} \quad \text{NextH} = \land \text{InputNum} \lor \text{Respond} \\
\quad \land \lor (\text{turn} = \text{"input"}) \land (h' = h) \\
\quad \lor (\text{turn} = \text{"output"}) \land (h' = h \cup \{x\})
\]

\[
\text{BY} \quad \text{DEF} \quad \text{NextH, Next, InputNumH, RespondH, InputNum, Respond}
\]

To define what a subaction is, we introduce the concept of a \textit{disjunctive representation}. A disjunctive representation of a formula \( N \) is a way of writing \( N \) in terms of subactions \( A_1, \ldots, A_m \) using only the operators \( \lor \) and \( \exists k \in K \), for some identifiers \( k \) and expressions \( K \). For example, consider the formula:

\[
(3.1) \quad B \lor C \lor D \lor (\exists i \in S, j \in T : \((\exists q \in U : E) \lor (\exists r \in W : F)\))
\]

where \( B, C, D, E, \) and \( F \) can be any formulas. Here is one of the 36 possible disjunctive representations of formula (3.1), where each boxed formula is a subaction:

\[
\boxed{B} \lor \boxed{C \lor D} \lor (\exists i \in S, j \in T : \boxed{(\exists q \in U : E) \lor (\exists r \in W : F)})
\]

In other words, this disjunctive representation of (3.1) has the four subactions \( B, C \lor D, \exists q \in U : E, \) and \( F \).

Each subaction of a disjunction representation has a \textit{context}, which is a pair \( \langle k; K \rangle \), where \( k \) is an \( n \)-tuple of identifiers and \( K \) is an \( n \)-tuple of expressions, for some \( n \). The contexts of the subactions in the disjunctive representation of (3.1) defined above are:

<table>
<thead>
<tr>
<th>subaction</th>
<th>context</th>
</tr>
</thead>
<tbody>
<tr>
<td>( B )</td>
<td>( \langle \rangle )</td>
</tr>
<tr>
<td>( C \lor D )</td>
<td>( \langle \rangle )</td>
</tr>
<tr>
<td>( \exists q \in U : E )</td>
<td>( \langle i, j; S, T \rangle )</td>
</tr>
<tr>
<td>( F )</td>
<td>( \langle i, j, r; S, T, W \rangle )</td>
</tr>
</tbody>
</table>

We let \( \exists \langle i, j; S, T \rangle \) be an abbreviation for \( \exists i \in S, j \in T \) and similarly for \( \forall \langle i, j; S, T \rangle \).

The generalization of this example should be clear. The tuple of identifiers in the context of a subaction \( A \) are all the bound identifiers of existential quantifiers
within whose scope $A$ lies.\footnote{Everything we do extends easily to handle unbounded quantification if we pretend that the unbounded quantifiers $\exists v$ and $\forall v$ are written $\exists v \in \Omega$ and $\forall v \in \Omega$, and we define $e \in \Omega$ to equal \textbf{true} for every expression $e$. Since unbounded quantification seldom occurs in specifications, we will not discuss this further.} If $\langle k; K \rangle$ is the empty context $\langle ; \rangle$, we let $\exists \langle k; K \rangle : F$ and $\forall \langle k; K \rangle : F$ equal $F$.

We can now define precisely what it means to add a history variable to a specification. The definition is contained in the hypothesis of this theorem:

**Theorem 1 (History Variable)** Let $Spec$ equal $Init \land \Box [Next]_{vars}$ and let $Spec^h$ equal $Init^h \land \Box [Next^h]_{vars^h}$, where:

- $Init^h$ equals $Init \land (h = \text{exp}_{Init})$, for some expression $\text{exp}_{Init}$ that may contain the specification’s (unprimed) variables.
- $Next^h$ is obtained from $Next$ by replacing each subaction $A$ of a disjunctive representation of $Next$ with $A \land (h' = \text{exp}_A)$, for some expression $\text{exp}_A$ that may contain primed and unprimed specification variables, identifiers in the context of $A$, and constant parameters.
- $vars^h$ equals $\langle vars, h \rangle$

Then $Spec$ is equivalent to $\exists h : Spec^h$.

The hypotheses of this theorem are purely syntactic ones: conditions on the definitions of $Init^h$, $Next^h$, and $vars^h$ plus conditions on what variables and identifiers may appear in $\text{exp}_{Init}$ and $\text{exp}_A$. By a variable or identifier appearing in an expression $exp$, we mean that it appears in the expression $\lfloor exp \rfloor$ obtained by expanding all definitions if $exp$. (See the discussion of definition expansion on page 7.)

### 3.3 Equivalence of Next-State Actions

When adding an auxiliary variable, it is often useful to rewrite a specification $Spec$—that is, to replace $Spec$ with a different but equivalent formula. This is most often done by rewriting the next-state action $Next$, which is done by rewriting one or more of the subactions in a disjunctive representation of $Next$. We now consider when we can replace a subaction $A$ in a disjunctive representation of $Next$ by the subaction $B$.

We can obviously replace $A$ by $B$ if $A$ and $B$ are equivalent formulas. However, this is often too stringent a requirement. For example, the two actions

\[
(x' = \text{IF } y \geq 0 \text{ THEN } x + y \text{ ELSE } x - y) \land (y' = y)
\]

\[
(x' = \text{IF } y > 0 \text{ THEN } x + y \text{ ELSE } x - y) \land (y' = y)
\]

are not equivalent. However, they are equivalent if $y$ is a number. Thus, we can replace one by the other in the next-state action if $y \in \text{Int}$ is an invariant of the specification. The generalization of this observation is:
Theorem 2 (Subaction Equivalence)  Let $A$ be a subaction with context $\langle k; K \rangle$ in a disjunctive representation of the next-state action of a specification $Spec$ with tuple vars of variables, let $Inv$ be an invariant of $Spec$, and let $B$ be an action satisfying:

\[(3.2) \quad Inv \Rightarrow \forall \langle k; K \rangle : A \equiv B\]

Then $Spec$ is equivalent to the specification obtained by replacing $A$ with $B$ in the next-state action’s disjunctive representation.

Formula (3.2) is an action formula, so it can be proved with TLAPS but cannot be checked with TLC.

TLC can check directly that two specifications are equivalent by checking that each specification implies the other. To check that specification $Spec$ implies a specification $SpecB$, just run TLC with a model having $Spec$ as the behavioral specification and $SpecB$ as the property to be checked. If one spec is obtained by a simple modification of the other, it should suffice to use small models. But in that case, it should not be hard to prove (3.2) with TLAPS, where $Inv$ is a simple type invariant.

3.4 Discussion of History Variables

As our example, we showed that specifications $MinMax1$ and $MinMax2$ are equivalent. In practice, we rarely care about checking equivalence of specifications. We almost always want to show that a specification $S$ satisfies some property $P$, which means that $S$ implies $P$. For example, $S \Rightarrow \Box Inv$ asserts that $Inv$ is an invariant of $S$. In (2.6) and (2.7), the property $P$ is, like $S$, a complete system specification.

The most general form of correctness that TLC can check is that one specification implies another. For example, the assertion that $Inv$ is an invariant of a specification $S$ with tuple $vars$ of variables is equivalent to the assertion that $S$ implies the specification

\[
Inv \land \Box[Inv' \equiv Inv]_{vars}
\]

We often want to show that a specification $S$ implies a higher-level, more abstract specification $T$. The standard way of doing this is to find a refinement mapping that expresses the values of $T$‘s variables as functions of the values of $S$‘s variables. This can’t be done if specification $T$ remembers in its state information that is forgotten by $S$. In that case, we show that $S$ implies $T$ by adding a history variable $h$ to $S$ to obtain the specification $S^h$, and we find a refinement mapping to show $S^h$ implies $T$. Since $\exists h : S^h$ is equivalent to $S$, this shows that $S$ implies $T$.

One can argue that $T$ is not a good higher-level specification if it keeps information about the past that doesn’t have to be kept by its implementation $S$. 

18
However, sometimes that information about the past can simplify the higher-level specification. We may also add a history variable to a specification $S$ so we can state the property we want to show that it satisfies, even if we aren’t explicitly constructing a refinement mapping. For example the property that $S$ requires one kind of action to occur before another can be expressed as an invariant if we add a history variable that remembers when actions have occurred. Because it’s a history variable, it doesn’t have to be implemented in the system being specified—that is, in the actual hardware or software. Only the variables of $S$ must be implemented.

3.5 Liveness

A natural liveness requirement for our MinMax specs is that every input should produce an output. This requirement is added to both specifications by conjoining the fairness requirement $WF_{vars}(Respond)$ to the formula $Spec$. (It is just a peculiarity of our example that the fairness requirements are exactly the same in both specifications.) Let us call the resulting specifications $LSpec$.

The two specifications are still equivalent when the internal variables are hidden. Formula (2.5), and hence formulas (2.6) and (2.7), remain true if we replace $Spec_1$ and $Spec_2$ by $LSpec_1$ and $LSpec_2$, respectively—where $LSpec_1$ and $LSpec_2$ are formulas $LSpec$ of MinMax1 and MinMax2, respectively. We verify (2.6) the same as before by verifying the theorem $LSpec \Rightarrow M!LSpec$ of module MinMax1. To verify (2.7), we need to add a history variable $h$ to $LSpec_2$ rather than to $Spec_2$. We now drop the subscripts; all the formulas we write will be ones defined in MinMax2 or MinMax2H.

To add a history variable to the specification $LSpec$, we add a history variable to its safety part and then conjoin the liveness part of $LSpec$. The resulting specification is defined in module MinMax2H by

$$HLSpec \triangleq HSpec \land WF_{vars}(Respond)$$

The equivalence of $LSpec$ and $\exists h : HLSpec$ follows from the equivalence of $Spec$ and $\exists h : HSpec$ by this argument:

1. $Spec \land WF_{vars}(Respond) \equiv (\exists h : HSpec) \land WF_{vars}(Respond)$
   
   **Proof:** Because $Spec$ is equivalent to $\exists h : HSpec$.

2. $(\exists h : HSpec) \land WF_{vars}(Respond)$
   
   $\equiv \exists h : (HSpec \land WF_{vars}(Respond))$

   **Proof:** For any formulas $F$ and $G$, if $h$ does not occur in $G$, then $(\exists h : F) \land G$ is equivalent to $\exists h : (F \land G)$.

3. Q.E.D.

   **Proof:** By definition of $LSpec$ and $HLSpec$, 1 and 2 imply that $LSpec$ is equivalent to $\exists h : HLSpec$. 

19
What we have done for this example generalizes in the obvious way. For a
specification written in the canonical form
\[
Init \land \Box[Next]_{vars} \land L
\]
with \( L \) a liveness condition, we add a history variable by adding it just to the
safety part, keeping the same liveness condition. This method of adding a history
variable to a specification with liveness produces unusual specifications. In our
example, if we expand the definition of \( HSpec \), we see that specification \( HLSpec \)
equals
\[
InitH \land \Box[NextH]_{varsH} \land WF_{varsH}(Respond)
\]
This spec is unusual because it contains a fairness condition on the action \( Respond \) that is not a subaction of the next-state relation \( NextH \). Such specs can
be weird. However, specs obtained in this way by adding a history variable are
not. Because (i) a \( Respond \) action is enabled iff a \( RespondH \) action is and (ii) a
\( NextH \) step is a \( Respond \) step iff it is a \( RespondH \) step, specification \( HLSpec \)
is equivalent to the normal specification:

\[
(3.3) \quad InitH \land \Box[NextH]_{varsH} \land WF_{varsH}(RespondH)
\]
In general, if \( Spec^h \) is obtained from a specification \( Spec \) by adding a history
variable \( h \), then replacing a fairness requirement on a subaction \( A \) by the same
fairness requirement on \( A^h \) produces a specification equivalent to \( Spec^h \).

The unusual nature of specification \( HLSpec \) affects neither the TLC model
checker nor our ability to reason about the specification. The same refinement
mapping as before shows that \( HLSpec \) implements \( MinMax1 \) with the added
fairness condition when internal variables are hidden.

4 Prophecy Variables

As we have observed, the fundamental task of verification is to show that the
specification \( Spec_1 \) of an implementation satisfies a specification \( Spec_2 \) of what
the implementation is supposed to do. A history variable remembers the past.
It is needed to find a refinement mapping to show that \( Spec_1 \) implements \( Spec_2 \)
when \( Spec_2 \) remembers previous events longer than it has to. A prophecy vari-
able is one that predicts the future. It is needed to find a refinement mapping
to show that \( Spec_1 \) implements \( Spec_2 \) when \( Spec_2 \) makes decisions before it has
to.

4.1 One-Prediction Prophecy Variables

We begin by showing how to add a simple prophecy variable that makes a single
prediction at a time. Suppose a disjunctive representation of the next-state
relation contains a subaction \( A \) such that

\[
(4.1) \quad A \Rightarrow (\exists i \in \Pi : Pred_A(i))
\]
for some expression $\text{Pred}_A(i)$ and constant set $\Pi$. Formula (4.1) is equivalent to

$$ (4.2) \quad A \equiv A \land (\exists i \in \Pi : \text{Pred}_A(i)) $$

which means that any $A$ step is an $A \land \text{Pred}_A(i)$ step for some $i$ in $\Pi$. We introduce a one-prediction prophecy variable $p$ whose value is an $i$ for which the next $A$ step is an $A \land \text{Pred}_A(i)$ step—if there is a next $A$ step. (There could be more than one such $i$, since we don’t require $\text{Pred}_A(i) \land \text{Pred}_A(j)$ to equal $\text{false}$ if $i \neq j$.) We give $p$ that meaning by replacing the subaction $A$ with a subaction $A^p$ defined by

$$ (4.3) \quad A^p \triangleq A \land \text{Pred}_A(p) \land \text{Setp} $$

where $\text{Setp}$ determines the value of $p'$.

To ensure that adding the prophecy variable $p$ allows all the behaviors of the other variables that the original spec does, we must ensure that $p$ can always have any value in $\Pi$. We do this by initializing $p$ to an arbitrary element of $\Pi$ and changing $p$ only by setting it to any arbitrary element of $\Pi$. Thus we modify the spec’s initial predicate $\text{Init}$ to equal $\text{Init} \land (p \in \Pi)$ and we let $\text{Setp}$ equal $p' \in \Pi$, so

$$ (4.4) \quad A^p \triangleq A \land \text{Pred}_A(p) \land (p' \in \Pi) $$

For another subaction $A$ of the next-state relation whose effect is not being predicted by $p$, we let $A^p$ leave the prediction unchanged, so it is defined simply as:

$$ (4.5) \quad A^p \triangleq A \land (p' = p) $$

We illustrate this with a simple example: a system in which integers are sent and received, where sending an integer $i$ is represented by setting the variable $x$ to $i$, and receiving a value is represented by setting $x$ to a value $\text{NotInt}$ that is not an integer. Our specification $\text{SendInt2}$ has the receiving action set an internal variable $z$ to the next value to be sent. (The initial value of $z$ is the first value to be sent.) This simple specification is in Figure 5.

Of course, we can describe the behavior of the variable $x$ even more simply, without using any internal variable. Such a specification is in module $\text{SendInt1}$ of Figure 6. Let $\text{Spec}_1$ and $\text{Spec}_2$ be the formulas $\text{Spec}$ of modules $\text{SendInt1}$ and $\text{SendInt2}$, respectively. It should be obvious that $\text{Spec}_1$ is equivalent to $\exists z : \text{Spec}_2$. To verify $(\exists z : \text{Spec}_2) \Rightarrow \text{Spec}_1$, we just have to verify $\text{Spec}_2 \Rightarrow \text{Spec}_1$, which TLC easily checks for a model that substitutes a finite set of integers for $\text{Int}$. We now show how to verify $\text{Spec}_1 \Rightarrow (\exists z : \text{Spec}_2)$.

There is obviously no refinement mapping under which $\text{Spec}_1$ implements $\text{Spec}_2$. Such a refinement mapping would be an expression $\bar{z}$ involving only the variable $x$ so that $\bar{z}$ is the value of $z$ in any state satisfying $\text{Spec}_2$. This is
module SendInt2

EXTENDS Integers

NotInt $\triangleq$ CHOOSE $n : n \not\in \text{Int}$

VARIABLE $x, z$

Init $\triangleq$ $\land x = \text{NotInt}$
       $\land z \in \text{Int}$

Send $\triangleq$ $\land x = \text{NotInt}$
       $\land x' = z$
       $\land z' = \text{NotInt}$

Rcv $\triangleq$ $\land x \in \text{Int}$
       $\land x' = \text{NotInt}$
       $\land z' \in \text{Int}$

Next $\triangleq$ Send $\lor$ Rcv

Spec $\triangleq$ Init $\land \square[\text{Next}]_{x, z}$

Figure 5: Specification SendInt2.

---

module SendInt1

EXTENDS Integers

NotInt $\triangleq$ CHOOSE $n : n \not\in \text{Int}$

VARIABLE $x$

Init $\triangleq$ $x = \text{NotInt}$

Send $\triangleq$ $\land x = \text{NotInt}$
       $\land x' \in \text{Int}$

Rcv $\triangleq$ $\land x \in \text{Int}$
       $\land x' = \text{NotInt}$

Next $\triangleq$ Send $\lor$ Rcv

Spec $\triangleq$ Init $\land \square[\text{Next}]_x$

Figure 6: Specification SendInt1
impossible, since \( z \) could equal any integer in a state in which \( x \) equals NotInt, so there is no way to express its value as a function of the value of \( x \). The variable \( z \) of \( SendInt2 \) is used to predict the value to be sent before it actually is sent. To be able to define the value of \( \bar{z} \) for a refinement mapping, we add a prophecy variable \( p \) to \( SendInt1 \) that predicts what the next value to be sent is.

The prophecy variable \( p \) must predict the value sent by action \( Send \) of \( SendInt1 \). Therefore \( SendP \) must have the form of (4.4). A little thought shows that \( p \) makes the right prediction if we take \( Pred_{Send}(p) \) to equal \( x' = p \). Since TLA\(^+\) doesn’t allow identifiers to have subscripts we write \( PredSend \) instead of \( Pred_{Send} \) and define

\[
PredSend(i) \triangleq x' = i
\]

Condition (4.1) becomes

\[
Send \Rightarrow \exists i \in \Pi : PredSend(i)
\]

which is obviously true by definition of \( Send \), if we let \( \Pi \) equal Int. Writing \( Pi \) instead of \( \Pi \) and \( SendP \) instead of \( Send^p \), we make the definitions:

\[
Pi \triangleq \text{Int}
\]

\[
SendP \triangleq Send \land PredSend(p) \land (p' \in Pi)
\]

(We could of course simply write \( Int \) instead of \( Pi \), but writing \( Pi \) will help us understand what’s going on.)

For the receive action, \( Rcv^p \) should have the form (4.5).

In a behavior of \( SendInt2 \), when \( x \) equals NotInt, the value of \( z \) is the next value sent; and when \( x \) equals an integer (the value sent), then \( z \) equals NotInt. Therefore, if \( SpecP \) is the specification obtained from specification \( Spec \) of \( SendInt1 \) by adding the prophecy variable \( p \), then \( SpecP \) implements the specification \( Spec \) of \( SendInt2 \) under this refinement mapping:

\[
\bar{z} \leftarrow \text{IF } x = \text{NotInt} \text{ THEN } p \text{ ELSE } \text{NotInt}
\]

Note that \( SpecP \) predicts the next value to be sent even before the \( SendInt2 \) specification does—when the previous value is sent rather than when it is received. Although it’s not necessary, we’ll see later how we could defer the prediction until the \( Rcv \) action is executed.

The complete specification is contained in module \( SendInt1P \) in Figure 7. Observe that we defined \( PredSend \) before the declaration of \( p \) to ensure that \( PredSend \) is not defined in terms of \( p \).

The module’s theorem asserts that \( SpecP \) implements formula \( Spec \) of \( SendInt2 \) under the refinement mapping defined above. It can be checked with TLC by creating a model with temporal specification \( SpecP \) and having it check the temporal property \( SI2!Spec \). The model will have to substitute a finite set of integers for \( Int \). The specification is very simple and doesn’t depend on any properties of integers, so substituting a set with a few numbers will ensure that we didn’t make a mistake.
Module SendInt1P

Extends SendInt1

\[ Pi \triangleq Int \]

\[ \text{PredSend}(i) \triangleq x' = i \]

Variable \( p \)

\[ \text{varsP} \triangleq \langle x, p \rangle \]

\[ \text{InitP} \triangleq \text{Init} \land (p \in \Pi) \]

\[ \text{SendP} \triangleq \text{Send} \land \text{PredSend}(p) \land (p' \in \Pi) \]

\[ \text{RcvP} \triangleq \text{Rcv} \land (p' = p) \]

\[ \text{NextP} \triangleq \text{SendP} \lor \text{RcvP} \]

\[ \text{SpecP} \triangleq \text{InitP} \land \Box [\text{NextP}]_{\text{varsP}} \]

\[ SI2 \triangleq \text{INSTANCE SendInt2 with } z \leftarrow \text{IF } x = \text{NotInt} \text{ THEN } p \text{ ELSE } \text{NotInt} \]

Theorem \( \text{SpecP} \Rightarrow SI2!\text{Spec} \)

Figure 7: Specification SendInt1P

4.2 One-Prediction Prophecy Variables in General

We generalize our description of a one-prediction prophecy variable in two ways. First, we can allow a prophecy variable to make predictions about more than one action by replacing more than one subaction \( A \) of a disjunctive representation by an action \( A^p \) of the form (4.3). If we do this for subactions \( A_1, A_2, \ldots \), then the value of \( p \) makes a prediction about the next step to occur that is an \( A_1 \) or \( A_2 \) or \( \ldots \) step. We can express this a little more elegantly by generalizing (4.3) to allow \( \text{Setp} \) to depend on \( A \) and then letting each action \( A \) of the disjunctive representation be replaced with an action \( A^p \) defined by

\[ A^p \triangleq A \land \text{Pred}_A(p) \land \text{Setp}_A \] (4.6)

For an action \( A \) about which no prediction is being made, \( \text{Pred}_A(p) \) is the expression TRUE. We can then replace (4.4) and (4.5) by defining \( \text{Setp}_A \) to be one of the following:

\[ (4.7) \quad \text{Setp}_A \triangleq p' = p \]
\[ \text{(b) } \text{Setp}_A \triangleq p' \in \Pi \]
where possibility (a) is allowed only if \( \text{Pred}_A(p) \) is the expression \( \text{true} \) (so \( p \) is making no prediction about \( A \)). This is more general because it allows an action that doesn’t use the prediction made by \( p \) to make a new prediction.

Our second generalization is needed to handle subactions of a disjunctive representation having a nonempty context. For a subaction \( A \) with context \( \langle k; K \rangle \), condition (4.1) contains the identifiers \( k \). That condition need only hold for values of those identifiers in the corresponding set in \( K \). Thus (4.1) can be generalized to

\[
(4.8) \quad \forall \langle k; K \rangle : A \Rightarrow (\exists i \in \Pi : \text{Pred}_A(i))
\]

Condition (4.8) is a condition on pairs of states. It needn’t hold for all pairs of states, only for pairs of states that can occur in a behavior satisfying the original specification \( \text{Spec} \). We can therefore replace (4.8) by the requirement\(^5\)

\[
(4.9) \quad \text{Spec} \Rightarrow \Box [\forall \langle k; K \rangle : A \Rightarrow (\exists i \in \Pi : \text{Pred}_A(i))]_{\text{vars}}
\]

TLC can check this condition with a model having the temporal formula \( \text{Spec} \) as its behavioral spec and

\[
\Box [\forall \langle k; K \rangle : A \Rightarrow (\exists i \in \Pi : \text{Pred}_A(i))]_{\text{vars}}
\]

as a property to be checked.

### 4.3 Prophecy Array Variables

Our next example is based on one created by Martín Abadi [1]. It is similar to our \( \text{SendInt} \) specifications in that a sender sends a value \( v \) to a receiver with a \( \text{Send} \) action that sets the variable \( x \) to \( v \), and the receiver receives the value by resetting \( x \). Instead of sending integers, the values sent are elements of an unspecified constant set \( \text{Data} \), and we let the initial value of \( x \) be a value \( \text{NonData} \) not in \( \text{Data} \). A variable \( y \) contains a set of values to be sent. Those values are chosen by a \( \text{Choose} \) action, which adds a new data element to \( y \). The high-level specification is formula \( \text{Spec} \) in module \( \text{SendSet} \) of Figure 8. We consider the variable \( x \) to be externally visible and \( y \) to be internal.

Our implementation adds to the specification of \( \text{SendSet} \) an \( \text{undo} \) operation that removes elements from \( y \). Abadi reports that this example is a highly simplified abstraction of a real system in which the implementation contains an \( \text{undo} \) operation not present in the specification.

The implementation specification is formula \( \text{SpecU} \) in module \( \text{SendSetUndo} \) of Figure 9. Its initial predicate is the same as the initial predicate of the specification \( \text{Spec} \) of module \( \text{SendSet} \), and its next-state action \( \text{NextU} \) is the same as the next-state action \( \text{Next} \) of that module except it allows \( \text{Undo}(S) \) steps that remove from \( y \) an arbitrarily chosen non-empty subset \( S \) of \( y \).

\(^5\)Formula (4.9) does not imply that (4.8) is true for stuttering steps that are allowed by \( A \). It can be shown that this doesn’t matter, and condition (4.9) is strong enough.
**MODULE** SendSet

**CONSTANT** Data

NonData \( \triangleq \) CHOOSE d : d \( \notin \) Data

**VARIABLES** x, y

vars \( \triangleq \) \( \langle x, y \rangle \)

Init \( \triangleq \) \( (x = \text{NonData}) \land (y = \{\}) \)

Choose \( \triangleq \) \( \land \exists d \in \text{Data} \setminus y : y' = y \cup \{d\} \land x' = x \)

Send \( \triangleq \) \( \land x = \text{NonData} \land x' \in y \land y' = y \setminus \{x'\} \)

Rcv \( \triangleq \) \( \land x \in \text{Data} \land x' = \text{NonData} \land y' = y \)

Next \( \triangleq \) Choose \( \lor \) Send \( \lor \) Rcv

Spec \( \triangleq \) Init \( \land \Box[\text{Next}]_{\text{vars}} \)

Figure 8: Specification SendSet.

**MODULE** SendSetUndo

**EXTENDS** SendSet

Undo(S) \( \triangleq \) \( \land y' = y \setminus S \land x' = x \)

NextU \( \triangleq \) Next \( \lor \) (\( \exists S \in (\text{subset} y) : \text{Undo}(S) \))

SpecU \( \triangleq \) Init \( \land \Box[\text{NextU}]_{\text{vars}} \)

Figure 9: Specification SendSetUndo.
It’s clear that the specifications $Spec$ of module $SendSet$ and $SpecU$ of module $SendSetUndo$ allow the same behaviors of the variable $x$. Hence, $\exists y : Spec$ and $\exists y : SpecU$ are equivalent. It’s easy to show that $Spec$ implements $SpecU$ under the identity refinement mapping in which $\bar{y}$ is defined to equal $y$, since $NextU$ allows all steps allowed by $Next$. This implies that $\exists y : Spec$ implies $\exists y : SpecU$. To construct a refinement mapping under which $SpecU$ implements $Spec$, we must define $y$ so that it contains a data value $d$ iff that value is going to be sent by a $Send$ step rather than being removed from $y$ by an $Undo$ step. This involves predicting, when $d$ is added to $y$, whether it will later be sent or “undone”.

We add a prophecy array variable $p$ to $SpecU$ that makes this prediction, setting $p[d]$ to either “send” or “undo” when $d$ is added to $y$. So, we define our set $Pi$ of possible predictions by

$$Pi \triangleq \{"send", "undo"\}$$

The value of $p$ in every state will be a function with domain equal to the set $y$, with $p[d] \in Pi$ for all $d \in y$. In other words $p \in [y \to Pi]$ will be an invariant of the spec $SpecUP$ obtained by adding the prophecy variable $p$. The variable $p$ is therefore making a predication $p[d]$ for every $d$ in $y$, so $p$ is making an array of prophecies. (This is a “dynamic” array, because the value of $y$ can change.)

We now define the specification $SpecUP$. As with a one-prediction prophecy variable, we obtain the next-state relation of $SpecUP$ by replacing each subaction $A$ in a disjunctive representation of the next-state relation with a new action $A^p$. Instead of defining $A^p$ as in (4.6), we define it to equal

$$A^p \triangleq A \land Pred_A(p) \land (p' \in NewPSet_A)$$

for suitable expressions $Pred_A(p)$ and $NewPSet_A$. We need a condition corresponding to condition (4.9) for a one-prediction prophecy variable to assert that there is a possible value of $p$ that makes $Pred_A(p)$ true. With an array prophecy variable, $p$ is no longer an element of $Pi$ but a function in $[Dom \to Pi]$ for some domain $Dom$ that can change. In our example, $Dom$ equals $y$. To make the generalization to an arbitrary spec easier, we define $Dom$ to equal $y$ and write $Dom$ instead of $y$ where appropriate. For our example, we can replace (4.9) with

$$\forall \langle k; K \rangle : A \Rightarrow (\exists f \in [Dom \to Pi] : Pred_A(f))_{\text{vars}}$$

where $\langle k; K \rangle$ is the context of $A$. (Remember that for the empty context $\langle ; ; \rangle$, we define $\forall \langle ; ; \rangle : F$ to equal $F$, for any formula $F$.)

We now define the formulas $Pred_A$ and $NewPSet_A$ for the disjunctive representation of $NextU$ with subactions $Choose$, $Send$, $Rcv$, and $Undo(S)$. The context of the first three subactions is empty; the context of $Undo(S)$ is $\langle S; \text{subset } y \rangle$.

The variable $p$ does not make any prediction about the $Choose$ action, so $Pred_{Choose}(p)$ should equal $\text{true}$. The action adds an element $d$ to its domain,
so \( \text{Choose}^p \) must allow \( p'[d] \) to equal any element of \( Pi \). For any element \( d \) in the domain \( Dom \) of \( p \), the value of \( p[d] \) can be left unchanged. Our definitions of \( \text{Pred}_{\text{Choose}}(p) \) and \( \text{NewPSet}_{\text{Choose}}(p) \) are then

\[
\text{Pred}_{\text{Choose}}(p) \triangleq \text{TRUE} \\
\text{NewPSet}_{\text{Choose}}(p) \triangleq \{ f \in [\text{Dom}' \rightarrow Pi] : \forall d \in \text{Dom} : f[d] = p[d] \}
\]

The prophecy variable \( p \) should predict that if the next action is a \( \text{Send} \) action, then it sends a value \( d \) in \( \text{Dom} \) such that \( p[d] = \text{"send"} \). The value sent by the action is \( x' \), so we define

\[
\text{Pred}_{\text{Send}}(p) \triangleq p[x'] = \text{"send"}
\]

The \( \text{Send} \) action removes the sent element \( d \) from \( \text{Dom} \), thus erasing the prediction \( p \) made about \( d \). The value of \( p[d] \) is left unchanged for all other elements \( d \) in \( \text{Dom} \). Thus \( \text{NewPSet}_{\text{Send}}(p) \) is defined as follows to be a set consisting of a single function

\[
\text{NewPSet}_{\text{Send}}(p) \triangleq \{ [d \in \text{Dom}' \mapsto p[d]] \}
\]

No prediction is made about the \( \text{Rcv} \) action, and it doesn’t change \( \text{Dom} \), so we have:

\[
\text{Pred}_{\text{Rcv}}(p) \triangleq \text{TRUE} \\
\text{NewPSet}_{\text{Rcv}}(p) \triangleq \{ p \}
\]

The \( \text{Undo}(S)^p \) action should be enabled only when \( p \) has predicted that all the elements in \( S \) will not be sent—in other words, when \( \forall d \in S : p[d] = \text{"undo"} \) is true. Since the identifier \( S \) appears in this formula, it must be an argument of the definition of \( \text{Undo}(S)^p \). Thus, we define

\[
\text{Pred}_{\text{Undo}}(p, S) \triangleq \forall d \in S : p[d] = \text{"undo"}
\]

The \( \text{Undo}(S) \) action removes from \( \text{Dom} \) all the elements for which \( p \) made a prediction about \( \text{Undo}(S) \)—namely, all the elements of \( S \). We can define \( \text{NewPSet}_{\text{Undo}(S)} \) the same way we defined \( \text{PSetSend}_{\text{Send}} \), without explicitly mentioning \( S \):

\[
\text{NewPSet}_{\text{Undo}}(p) \triangleq \{ [d \in \text{Dom}' \mapsto p[d]] \}
\]

We can now declare the variable \( p \) and define the specification \( \text{Spec}_{\text{UP}} \) by defining the initial predicate \( \text{Init}_{\text{UP}} \) and defining the next-state relation \( \text{Next}_{\text{UP}} \) in terms of the subactions \( A^p \), using (4.10). The complete specification is in module \( \text{SendSetUndoP} \) shown in Figure 10. Note that the initial value of \( p \) is the unique function whose domain is the empty set. We could write that function as \( [d \in \{ \} \mapsto \text{exp}] \) for any expression \( \text{exp} \)—for example, 42. However, it’s easier to write that function as \( \langle \rangle \) (the empty sequence).

We can now define the refinement mapping under which \( \text{Spec}_{\text{UP}} \) implements specification \( \text{Spec} \) of module \( \text{SendSet} \). The refinement mapping defines \( \overline{y} \) to equal the set of elements \( d \) in \( y \) with \( p[d] = \text{"send"} \). We thus add to the module
EXTENDS SendSetUndo

\[ Pi \triangleq \{ "send", "undo" \} \]
\[ Dom \triangleq y \]

\[ PredChoose(p) \triangleq \text{TRUE} \]
\[ NewPSetChoose(p) \triangleq \{ f \in [Dom' \rightarrow Pi] : \forall d \in Dom : f[d] = p[d] \} \]

\[ PredSend(p) \triangleq p[x'] = "send" \]
\[ NewPSetSend(p) \triangleq \{ [d \in Dom' \mapsto p[d]] \} \]

\[ PredRcv(p) \triangleq \text{TRUE} \]
\[ NewPSetRcv(p) \triangleq \{ p \} \]

\[ PredUndo(p, S) \triangleq \forall d \in S : p[d] = "undo" \]
\[ NewPSetUndo(p) \triangleq \{ [d \in Dom' \mapsto p[d]] \} \]

VARIABLE \( p \)
\[ \text{varsP} \triangleq \langle \text{vars}, p \rangle \]

\[ \text{InitUP} \triangleq \text{Init} \wedge (p = \langle \rangle) \]

\[ \text{ChooseP} \triangleq \text{Choose} \wedge \text{PredChoose}(p) \wedge (p' \in \text{NewPSetChoose}(p)) \]
\[ \text{SendP} \triangleq \text{Send} \wedge \text{PredSend}(p) \wedge (p' \in \text{NewPSetSend}(p)) \]
\[ \text{RcvP} \triangleq \text{Rcv} \wedge \text{PredRcv}(p) \wedge (p' \in \text{NewPSetRcv}(p)) \]
\[ \text{UndoP}(S) \triangleq \text{Undo}(S) \wedge \text{PredUndo}(p, S) \wedge (p' \in \text{NewPSetUndo}(p)) \]

\[ \text{NextUP} \triangleq \text{ChooseP} \lor \text{SendP} \lor \text{RcvP} \lor (\exists S \in \text{SUBSET} y : \text{UndoP}(S)) \]

\[ \text{SpecUP} \triangleq \text{InitUP} \wedge \Box[\text{NextUP}]_{\text{varsP}} \]

Figure 10: Specification \( \text{SendSetUndoP} \)

\[ SS \triangleq \text{INSTANCE} \text{SendSet} \text{ WITH} y \leftarrow \{ d \in y : p[d] = "send" \} \]

THEOREM \( \text{SpecUP} \Rightarrow SS!Spec \)

We can have TLC check this theorem by creating a model having \( \text{SpecUP} \) as the behavior spec and checking the property \( SS!Spec. \)

We should also check that condition (4.11) holds for each subaction \( A. \) To do this, we need to create a model with specification \( \text{SpecU} \) and have TLC check the property:

\[ \Box[ \wedge \text{Choose} \Rightarrow \exists f \in [\text{Dom} \rightarrow Pi] : \text{PredChoose}(f) \]
\[ \wedge \text{Send} \Rightarrow \exists f \in [\text{Dom} \rightarrow Pi] : \text{PredSend}(f) \]
\[ \wedge \text{Rcv} \Rightarrow \exists f \in [\text{Dom} \rightarrow Pi] : \text{PredRcv}(f) \]
\[ \forall S \in \text{SUBSET } y : \]
\[ \text{Undo}(S) \Rightarrow \exists f \in [\text{Dom} \rightarrow Pi] : \text{PredUndo}(f, S) \]

However, there is a problem in doing this. TLC will not allow a model for module \text{SendSetUndoP} to have behavior specification \text{SpecU} because that spec doesn’t describe the behavior of the variable \( p \). We can solve this problem by modifying the specification—temporarily inserting “======” into the module before the declaration of \( p \) and then creating the necessary model. Alternatively, we can move all the definitions before the declaration of \( p \) into module \text{SendSetUndo} and check the condition in a model for that spec. This is inelegant because those definitions aren’t part of the \text{SendSetUndo} specification. The proper solution is to move those definitions from \text{SendSetUndoP} and put them in a new module that extends \text{SendSetUndo} and is extended by \text{SendSetUndoP}. We can then check that the condition is satisfied using a model for that specification that has behavior spec \text{SpecU}. We won’t bother doing this, instead putting them in module \text{SendSetUndoP}.

Recall that in the \text{SendInt} example of Section 4.1, the one-prediction prophecy variable we used predicted the next value to be sent when the previous value was sent, while the specification \text{SendInt2} didn’t choose the next value to be sent until the previous value was received. We can use an array prophecy variable to defer the prediction until it’s needed. We let the domain \text{Dom} of \( p \) initially contain a single element—let’s take that element to be “on”. We let the \text{Send} action set \text{Dom} to the empty set, and we let the \text{Rcv} action set \text{Dom} to \{“on”\}.

### 4.4 Prophecy Data Structure Variables

It is easy to generalize from the \text{SendSet} example to an arbitrary prophecy-array variable. However, it is useful to generalize still further from an array to an arbitrary data structure. These prophecy data structure variables are the most general ones that we consider. We propose them as the standard way of defining prophecy variables in TLA\(^+\), and we have created a module with definitions that simplify adding these prophecy variables. Both single-prophecy variables and prophecy array variables are easily expressed as special cases.

As our example of a prophecy data structure variable, we modify the specification \text{SendSet}, in which a set of items to be sent is chosen, to a specification \text{SendSeq} in which a sequence of items is chosen. The value of the variable \( y \) is changed from a set of data items to a sequence of data items. The next item to be sent is the one at the head of \( y \), and each value chosen is appended to the tail of \( y \). The specification is in module \text{SendSeq}, shown in Figure 11.

For our implementation, we add an undo action that removes an arbitrary element from the sequence \( y \). The specification is in module \text{SendSeqUndo} of
EXTENDS Sequences, Integers

CONSTANT Data

NonData \triangleq \text{choose } v : v \notin Data

VARIABLES x, y
vars \triangleq \langle x, y \rangle

Init \triangleq (x = \text{NonData}) \land (y = \langle \rangle)

Choose \triangleq \land \exists d \in Data : y' = \text{Append}(y, d)
\land x' = x

Send \triangleq \land x = \text{NonData} \land y \neq \langle \rangle
\land x' = \text{Head}(y)
\land y' = \text{Tail}(y)

Rcv \triangleq \land x \in Data
\land x' = \text{NonData}
\land y' = y

Next \triangleq \text{Choose} \lor \text{Send} \lor \text{Rcv}

Spec \triangleq \text{Init} \land \Box[\text{Next}]_\text{vars}

Figure 11: Specification SendSeq

Figure 12. It defines $\text{RemoveEltFrom}(i, seq)$ to be the sequence obtained from a sequence $seq$ by removing its $i^{th}$ element, assuming $1 \leq i \leq \text{Len}(seq)$.

As before, we want to show that specification $\text{SendU}$ of module $\text{SendSeqUndo}$ implements $\exists \ y : \text{Spec}$, where $\text{Spec}$ is the specification of module $\text{SendSeq}$. Again, we need to add a prophecy variable $p$ that predicts whether each element of $y$ will be sent or “undone”. We do this by having $p$ be an element of $\text{Seq}\{\text{“send”, “undo”}\}$ that has the same length as $y$. The $\text{Choose}^p$ action should append either “send” or “undo” to the tail of $p$, the $\text{Send}^p$ action should remove the head of $p$, and the $\text{Undo}(i)$ action should remove the $i^{th}$ element of $p$.

As we did for the SendSet example, we write a module $\text{SendSeqUndoP}$ that extends $\text{SendSeqUndo}$. In it, we define the formulas $\text{Pred}_A(p)$ of (4.10) for each of the subactions Choose, Send, Rcv, and Undo$(i)$. The definitions are:

\begin{align*}
\text{PredChoose}(p) & \triangleq \text{TRUE} \\
\text{PredSend}(p) & \triangleq p[1] = \text{“send”} \\
\text{PredRcv}(p) & \triangleq \text{TRUE} \\
\text{PredUndo}(p, i) & \triangleq p[i] = \text{“undo”}
\end{align*}
We now need to define the expression $NewPSet_A$ of (4.10) for each of these subactions.

Since a sequence of length $n$ is a function with domain $1..n$, the value of $p$ is a function—just as in $SendSetUndoP$ above. This time its domain $Dom$ is the set $1..Len(y)$. However, in that example, if $d$ is in the domain $Dom$ of $p$ in two successive states, then $p[d]$ represents the same prediction in both states. This isn’t true in the current example. If $s \rightarrow t$ is a $Send$ step and $Len(p) > 1$ in state $s$, then the prediction made by $p[2]$ in state $s$ is the prediction made by $p[1]$ in state $t$. If $s \rightarrow t$ is an $Undo(i)$ step and $j > i$, the prediction made by $p[j]$ in state $s$ is the prediction made by $p[j-1]$ in state $t$.

In general, an action $A$ defines a correspondence between some elements in the domain $Dom$ of $p$ and elements in the domain $Dom'$ of $p'$. For example, for the $Send$ action, each element $i > 1$ in $Dom$ corresponds to the element $i-1$ of $Dom'$. The formula $p' \in NewPSet_A$ in (4.10) should ensure that if an element $d$ of $Dom'$ either corresponds to an element $c$ of $Dom$ that makes a prediction about $A$ or else does not correspond to any element of $Dom$, then $p'[d]$ can assume any value in $\Pi$; but if $d$ corresponds to an element $c$ of $Dom$ that make no prediction about $A$, then $p'[d]$ equals $p[c]$. Instead of defining the formulas $NewPSet_A$ directly, we will define them in terms of the correspondence between elements of $Dom$ and $Dom'$ made by $A$ and the set of elements $d$ in $Dom$ for which $p[d]$ makes a prediction about $A$.

To express formally a correspondence between elements of $Dom$ and $Dom'$, we introduce the concept of a partial injection. A partial function from a set $U$ to a set $V$ is a function from a subset of $U$ to $V$. In other words, it is an element of $[D \rightarrow V]$ for some subset $D$ of $U$. (Remember that $U$ is a subset of itself.) An injection is a function that maps different elements in its domain to different values. In other words, a function $f$ is an injection iff for all $a$ and $b$ in domain $f$, if $a \neq b$ then $f[a] \neq f[b]$. The set of all partial injections from $U$ to
V is defined in TLA+ by

\[ \text{PartialInjections}(U, V) \triangleq \]
\[ \text{let } \text{PartialFcns} \triangleq \text{union } \{ [D \to V] : D \in \text{subset } U \} \]
\[ \text{in } \{ f \in \text{PartialFcns} : \forall a, b \in \text{domain } f : (a \neq b) \Rightarrow (f[a] \neq f[b]) \} \]

For each subaction A, we define a partial injection \( \text{DomInj}_A \) from \( \text{Dom} \) to \( \text{Dom}' \) such that an element \( c \) of \( \text{Dom} \) corresponds to an element \( d \) of \( \text{Dom}' \) iff \( c \) is in the domain of \( \text{DomInj}_A \) and \( d = \text{DomInj}_A[c] \). Here are the definitions for the four subactions, which are put in module \( \text{SendSeqUndoP} \):

\[ \text{DomInjChoose} \triangleq \{ d \in \text{Dom} \mapsto d \} \]
\[ \text{DomInjSend} \triangleq \{ i \in 2..\text{Len}(y) \mapsto i - 1 \} \]
\[ \text{DomInjRcv} \triangleq \{ d \in \text{Dom} \mapsto d \} \]
\[ \text{DomInjUndo}(i) \triangleq \{ j \in 1..\text{Len}(y) \setminus \{i\} \mapsto \text{if } j < i \text{ THEN } j \text{ ELSE } j - 1 \} \]

For the prophecy array variable described in Section 4.3 above, the function \( \text{DomInj}_A \) maps each element \( d \) in \( \text{Dom} \) that is also in \( \text{Dom}' \) to itself. Thus, for each subaction \( A \) used in defining a prophecy array variable, we can define

\[ \text{DomInj}_A \triangleq \{ d \in \text{Dom} \cap \text{Dom}' \mapsto d \} \]

A function \( f \) such that \( f[x] = x \) for all \( x \in \text{domain } f \) is called an identity function. For convenience, the \( \text{Prophecy} \) module defines \( \text{IdFcn}(S) \) to be the identity function with domain \( S \). But we won’t bother using it here. The \( \text{Prophecy} \) module also defines \( \text{EmptyFcn} \) to be the (unique) function whose domain is the empty set.

Let us return to our prophecy data structure example. For a subaction \( A \), we can define \( \text{NewPSet}_A \) in terms of \( \text{DomInj}_A \) and the subset \( \text{PredDom}_A \) of \( \text{Dom} \), which consists of the elements in \( \text{Dom} \) such that \( p[d] \) makes a prediction about \( A \). We define these sets \( \text{PredDom}_A \) in module \( \text{SendSeqUndoP} \) for our four subactions as follows:

\[ \text{PredDomChoose} \triangleq \{ \} \]
\[ \text{PredDomSend} \triangleq \{ 1 \} \]
\[ \text{PredDomRcv} \triangleq \{ \} \]
\[ \text{PredDomUndo}(i) \triangleq \{ i \} \]

(Since \( \text{PredDomUndo}(i) \) depends on the identifier \( i \) in its context, we must define \( \text{PredDomUndo} \) to have a parameter.)

We can define \( \text{NewPSet}_A \) to equal the set of all functions \( q \) in \([\text{Dom'} \to \Pi]\) such that for every element \( d \) in \( \text{Dom} \) that is not in \( \text{PredDom}_A \) and has a corresponding element \( \text{DomInj}_A[d] \) in \( \text{Dom}' \), the value of \( q \) on that corresponding element equals \( p[d] \). More precisely, \( \text{NewPSet}_A \) equals:

\[ \{ q \in [\text{Dom'} \to \Pi] : \forall d \in (\text{domain } \text{DomInj}_A) \setminus \text{PredDom}_A : q[\text{DomInj}_A[d]] = p[d] \} \]
We encapsulate definitions like this in a module Prophecy. We find it most convenient to make this a constant module with constant parameters \( \Pi \), \( \text{Dom} \), and \( \text{DomPrime} \). This module is meant to be instantiated with the parameter \( \Pi \) instantiated by \( \Pi \), with the parameter \( \text{Dom} \) instantiated by the appropriate state function \( \text{Dom} \), and with \( \text{DomPrime} \) instantiated by \( \text{Dom}' \). The following definition from module Prophecy allows us to define \( \text{NewPSet}_A(p) \) to equal \( \text{NewPSet}(p, \text{DomInj}_A, \text{PredDom}_A) \):

\[
\text{NewPSet}(p, \text{DomInj}, \text{PredDom}) \triangleq \\
\{ q \in [\text{DomPrime} \to \Pi] : \\
\forall d \in (\text{DOMAIN} \text{DomInj}) \setminus \text{PredDom} : q[\text{DomInj}[d]] = p[d] \}
\]

For each action in \( A \) in our disjunctive decomposition of NextU, we have written the definitions of \( \text{Pred}_A \), \( \text{DomInj}_A \), and \( \text{PredDom}_A \). This allows us to define \( \text{NewPSet}_A \), and therefore, by (4.10), to define \( A^p \). The following operator \( \text{ProphAction} \) from the Prophecy module allows us to write \( A^p \) as \( \text{ProphAction}(A, p, p', \text{DomInj}_A, \text{PredDom}_A) \):

\[
\text{ProphAction}(A, p, p', \text{DomInj}, \text{PredDom}, \text{Pred}(-)) \triangleq \\
A \wedge \text{Pred}(p) \wedge (p' \in \text{NewPSet}(p, \text{DomInj}, \text{PredDom}))
\]

In module SendSeqUndoP, we can define

\[
\text{ChooseP} \triangleq \text{ProphAction}(\text{Choose}, p, p', \text{DomInjChoose}, \text{PredDomChoose}, \text{PredChoose})
\]

The definitions of SendP and RcvP are similar. However, there is a problem with the definition of \( \text{Undo}(i)^p \), which we write as \( \text{UndoP}(i) \). Operator \( \text{ProphAction} \) requires its last argument, which represents \( \text{Pred}_A \), to be an operator with a single argument. However, we defined \( \text{PredUndo} \) to have two arguments: \( p \) and its context identifier \( i \). Since we are defining \( \text{UndoP}(i) \), the fifth argument has to be an operator \( \text{Op} \) so that \( \text{Op}(p) \) equals \( \text{PredUndo}(p, i) \). So, we should define:

\[
\text{UndoP}(i) \triangleq \text{LET } \text{Op}(j) \triangleq \text{PredUndo}(j, i) \\\n\text{IN } \text{ProphAction}(\text{Undo}(i), p, p', \text{DomInjUndo}(i), \text{PredDomUndo}(i), \text{Op})
\]

Using the TLA\(^+\) lambda construct (added since Specifying Systems was published), this can also be written as:

\[
\text{UndoP}(i) \triangleq \\
\text{ProphAction}( \text{Undo}(i), p, p', \text{DomInjUndo}(i), \text{PredDomUndo}(i), \text{LAMBDA } j : \text{PredUndo}(j, i))
\]

It’s now straightforward to complete our definition of specification SpecUP. Doing so, gathering up the definitions made or implied so far, and rearranging them a bit, we get the beginning of module SendSeqUndoP shown in Figure 13.
module SendSeqUndoP

EXTENDS SendSeqUndo

Pi \triangleq \{ "send", "undo" \}
Dom \triangleq \text{DOMAIN } y

INSTANCE Prophecy WITH DomPrime \leftarrow Dom'

PredDomChoose \triangleq \{}
DomInjChoose \triangleq [d \in \text{Dom} \mapsto d]
PredChoose(p) \triangleq \text{TRUE}

PredDomSend \triangleq \{1\}
DomInjSend \triangleq [i \in 2 \ldots \text{Len}(y) \mapsto i - 1]
PredSend(p) \triangleq p[1] = "send"

PredDomRcv \triangleq \{}
DomInjRcv \triangleq [d \in \text{Dom} \mapsto d]
PredRcv(p) \triangleq \text{TRUE}

PredDomUndo(i) \triangleq \{i\}
DomInjUndo(i) \triangleq [j \in 1 \ldots \text{Len}(y) \setminus \{i\} \mapsto \text{if } j < i \text{ THEN } j \text{ ELSE } j - 1]
PredUndo(p, i) \triangleq p[i] = "undo"

VARIABLE p
varsP \triangleq ⟨vars, p⟩

InitUP \triangleq \text{Init} \land (p \in [\text{Dom} \rightarrow Pi])

ChooseP \triangleq \text{ProphAction(Choose, p, p’,}
\quad \text{DomInjChoose, PredDomChoose, PredChoose)}

SendP \triangleq \text{ProphAction(Send, p, p’, DomInjSend, PredDomSend, PredSend)}

RcvP \triangleq \text{ProphAction(Rcv, p, p’, DomInjRcv, PredDomRcv, PredRcv)}

UndoP(i) \triangleq \text{ProphAction(undo(i), p, p’, DomInjUndo(i), PredDomUndo(i),}
\quad \text{LAMBDA } j : PredUndo(j, i))

NextUP \triangleq \text{ChooseP} \lor \text{SendP} \lor \text{RcvP} \lor (\exists i \in 1 \ldots \text{Len}(y) : UndoP(i))

SpecUP \triangleq \text{InitUP} \land \square[\text{NextUP}]_{varsP}

Figure 13: The specification of SpecUP.
Finally, we have to define the refinement mapping under which \( \text{SpecUP} \) implements specification \( \text{Spec} \) of module \( \text{SendSeq} \). The idea is simple: we let \( \overline{y} \) be the subsequence of \( y \) containing only those elements for which the corresponding element of the sequence \( p \) equals “send”. The following formal definition is a bit tricky. It uses a local recursive definition of an operator \( R \) such that if \( y_{\text{seq}} \) is any sequence and \( p_{\text{seq}} \) is a sequence of the same length, then \( R(y_{\text{seq}}, p_{\text{seq}}) \) is the subsequence of \( y_{\text{seq}} \) that contains \( y_{\text{seq}}[i] \) iff \( p_{\text{seq}}[i] \) equals “send”.

\[
\begin{align*}
\overline{y} & \triangleq \\
& \text{let recursive } R(-, -) \\
R(y_{\text{seq}}, p_{\text{seq}}) & \triangleq \\
& \text{if } y_{\text{seq}} = \langle \rangle \\
& \text{then } y_{\text{seq}} \\
& \text{else if } \text{Head}(p_{\text{seq}}) = \text{"send"} \\
& \text{then } \langle \text{Head}(y_{\text{seq}}) \rangle \circ R(\text{Tail}(y_{\text{seq}}), \text{Tail}(p_{\text{seq}})) \\
& \text{else } R(\text{Tail}(y_{\text{seq}}), \text{Tail}(p_{\text{seq}})) \\
\in R(y, p)
\end{align*}
\]

We then instantiate module \( \text{SendSeq} \) and state as follows the theorem asserting that \( \text{SpecUP} \) implements formula \( \text{Spec} \) of that model under the refinement mapping.

\[
\text{SS} \triangleq \text{instance } \text{SendSeq} \text{ with } y \leftarrow \overline{y}
\]

**Theorem**: \( \text{SpecUP} \Rightarrow \text{SS} \vdash \text{Spec} \)

TLC can check this theorem in the usual way.

### 4.5 Checking the Definitions

We have shown how to define a specification \( \text{Spec}^p \) for an arbitrary specification \( \text{Spec} \) by defining a state function \( \text{Dom} \) and, for every subaction \( A \) of a disjunctive representation of the next-state action of \( \text{Spec} \), defining \( \text{Pred}_A \), \( \text{DomInj}_A \), and \( \text{PredDom}_A \). These definitions must satisfy certain conditions to ensure that \( \exists p : \text{Spec}^p \) is equivalent to \( \text{Spec} \). We now state those conditions.

The first condition is (4.11). The \( \text{Prophecy} \) module defines this operator:

\[
\text{ExistsGoodProphecy}(\text{Pred}(\_)) \triangleq \exists q \in [\text{Dom} \rightarrow \Pi] : \text{Pred}(q)
\]

For a subaction \( A \) with an empty context, we can write (4.11) as

\[
\text{Spec} \Rightarrow \Box[A \Rightarrow (\text{ExistsGoodProphecy}(\text{Pred}_A))_{\text{vars}}
\]

(Remember that the \( \text{Prophecy} \) module will be instantiated with the appropriate expression substituted for \( \text{Dom} \).) To see how this definition is used if \( A \) has a non-empty context, here is how condition is (4.11) is expressed for the subaction \( \text{UndoP}(i) \) of specification \( \text{SpecU} \) in our \( \text{SendSeq} \) example:
\begin{align*}
\text{SpecU} & \Rightarrow \quad \Box \left[ \forall i \in \text{Dom} : \right. \\
& \quad \quad \quad \quad \text{Undo}(i) \Rightarrow \\
& \quad \quad \quad \quad \quad \quad \exists \text{GoodProphecy}(\lambda p : \text{PredUndo}(p, i)) \left] \right. \text{vars}
\end{align*}

The only condition we require of \text{DomInj}_A is that it be a partial function from \text{Dom} to \text{Dom}'. This is expressed as \text{IsDomInj}(\text{DomInj}_A) using this definition from module \text{Prophecy}

$$\text{IsDomInj}(\text{DomInj}) \triangleq \text{DomInj} \in \text{PartialInjections}(\text{Dom}, \text{DomPrime})$$

As with the \text{ExistsGoodProphecy} condition, it needs to hold only for \text{A} steps in a behavior satisfying the specification \text{Spec}. Hence the general requirement on \text{DomInj}_A for an action \text{A} with context \text{⟨k; K⟩} is

$$\text{Spec} \Rightarrow \quad \Box \left[ \forall \langle k; K \rangle : \text{A} \Rightarrow \text{IsDomInj}(\text{DomInj}_A) \right] \text{vars}$$

Because \text{IsDomInj} does not have an operator argument, no local definition or \text{LAMBDA} expression is needed even if the context is nonempty. For example, if \text{A} is subaction \text{Undo}(i) of specification \text{SpecU} of the \text{SendSeq} example, this condition is written:

$$\forall i \in \text{Dom} : \text{Undo}(i) \Rightarrow \text{IsDomInj}(\text{DomInj}_{\text{Undo}}(i))$$

Finally, we need a condition on \text{PredDom}_A. Remember that \text{PredDom}_A should equal the set of elements \text{d} of \text{Dom} such that \text{p}[\text{d}] is making predictions about \text{A}. Actually, it suffices that \text{PredDom}_A contain all such elements. (It may contain other elements as well.) This is equivalent to the requirement that any element not in \text{PredDom}_A does not make a prediction about \text{A}. Making a prediction about \text{A} means affecting the value of \text{Pred}_A, so not making a prediction means not affecting its value. Thus, \text{p}[\text{d}] does not make a prediction about \text{A} iff setting \text{p}[\text{d}] to any value in \Pi does not change the value of \text{Pred}_A. You should be able to convince yourself that the value of \text{Pred}_A does not depend on the value of \text{p}[\text{d}] for any \text{d} not in \text{PredDom}_A iff the following formula is true:

$$\forall q, r \in [\text{Dom} \to \Pi] : \\
(\forall d \in \text{PredDom}_A : q[d] = r[d]) \Rightarrow (\text{Pred}_A(q) = \text{Pred}_A(r))$$

In addition to this requirement, to ensure that our formulas make sense, we also make the obvious requirement that \text{PredDom}_A is a subset of \text{Dom}. The following definition appears in module \text{Prophecy}.

$$\text{IsPredDom}(\text{PredDom}, \text{Pred}(_-)) \triangleq \\
\wedge \text{PredDom} \subseteq \text{Dom} \\
\wedge \forall q, r \in [\text{Dom} \to \Pi] : \\
(\forall d \in \text{PredDom} : q[d] = r[d]) \Rightarrow (\text{Pred}(q) = \text{Pred}(r))$$
\[
\text{Condition} \triangleq \\
\land \text{ProphCondition}(\text{Choose}, \ DomInj\text{Choose}, \ PredDom\text{Choose}, \ Pred\text{Choose}) \\
\land \text{ProphCondition}(\text{Send}, \ DomInj\text{Send}, \ PredDom\text{Send}, \ Pred\text{Send}) \\
\land \text{ProphCondition}(Rcv, \ DomInj\text{Rcv}, \ PredDom\text{Rcv}, \ Pred\text{Rcv}) \\
\land \forall i \in \text{Dom} : \\
\text{ProphCondition}(\text{Undo}(i), \ DomInj\text{Undo}(i), \ PredDom\text{Undo}(i), \\
\text{LAMBDA } p : \text{PredUndo}(p, i))
\]

**THEOREM** \( \text{Spec}_U \Rightarrow \Box [\text{Condition}]_{\text{vars}} \)

Figure 14: Action requirements for specification \( \text{SendSeqUndo} \).

Remembering that the condition on \( \text{PredDom}_A \) needs to hold only for \( A \) steps in a behavior satisfying the specification, we can express it for an action \( A \) with an empty context as:

\[
\text{Spec} \Rightarrow \Box [A \Rightarrow \text{IsPredDom}(\text{PredDom}_A, \text{Pred}_A)]_{\text{vars}}
\]

Because the second argument of \( \text{IsPredDom} \) is an operator argument, we again need to use a local definition or a \text{LAMBDA} expression to express the condition if the subaction \( A \) has a nonempty context. For example, here is how we can express it for the \( \text{Undo}(i) \) subaction in our \( \text{SendSeq} \) example using a local definition:

\[
\Box [\forall i \in \text{Dom} : \\
\text{Undo}(i) \Rightarrow \text{LET } Op(p) \triangleq \text{PredUndo}(p, i) \\
\text{IN } \text{IsPredDom}(\text{PredDom\text{Undo}}, \text{Op})]_{\text{vars}}
\]

Here is how it is written with a \text{LAMBDA} expression:

\[
\Box [\forall i \in \text{Dom} : \\
\text{Undo}(i) \Rightarrow \text{IsPredDom}(\text{PredDom\text{Undo}}, \\
\text{LAMBDA } p : \text{PredUndo}(p, i))]_{\text{vars}}
\]

The following definition from module \( \text{Prophecy} \) allows us to combine the three conditions.

\[
\text{ProphCondition}(A, \ Dom\text{Inj}, \ Pred\text{Dom}, \ Pred(_)) \triangleq \\
A \Rightarrow \land \text{ExistsGoodProphecy}(\text{Pred}) \\
\land \text{IsDomInj}(\text{Dom\text{Inj}}) \\
\land \text{IsPredDom}(\text{Pred\text{Dom}}, \text{Pred})
\]

Using this definition, the requirements on the definitions are expressed for specification \( \text{SendSeqUndo} \) in Figure 14.

We encounter the same problem here that we encountered in checking condition (4.11) for the \( \text{SendSet} \) example. We would like to put these requirements
in module \textit{SendSetUndoP} (Figure 10), right before the declaration of the variable \( p \). However, TLC can’t check the theorem in a model for that specification because \( SpecU \) does not specify the values of variable \( p \). We can either move all the definitions that are now in \textit{SendSetUndoP} before the declaration of \( p \) into a separate module, put them at the end of module \textit{SendSetUndo}, or end the module before the declaration of \( p \) by adding “========” when checking the condition.

You should check these conditions when adding a prophecy variable. They provide a good way to debug your definitions, before you try checking that the specification with prophecy variable implements the desired specification.

4.6 Liveness

As with our other auxiliary variables, we add a prophecy variable to the safety part of a specification, but we keep the liveness part the same. As we remarked in Section 3.5, this produces unusual specifications in which the liveness property can assert a fairness condition about an action that isn’t a subaction of the next-state action.

For history variables, although the form of the specifications is unusual, the specifications are not. This is not the case for prophecy variables. If \( Spec \) has a liveness condition, the specification \( Spec^p \) obtained from it by adding a prophecy variable can be weird. As an example, suppose that in the \textit{SendInt} specifications of Section 4.1, instead of taking \( Pi \) to equal \( Int \), we let it equal \( Int \cup \{\infty\} \) for some value \( \infty \notin Int \). Everything we did would work exactly as before, and the theorem at the end of module \textit{SendInt1P} in Figure 7 would still be true. If a \textit{SendP} step set \( p' \) to \( \infty \), predicting that the next value to be sent is \( \infty \), then the system would simply halt (stutter forever) before the next \textit{SendP} step because \( Send \land \text{PredSend}(\infty) \) equals FALSE.

Now suppose we add a liveness condition \( WF_{vars}(Next) \) to our \textit{SendInt} specifications, requiring that they never halt. We would then have

\[
SpecP \triangleq \text{InitP} \land \Box[NextP]_{varsP} \land WF_{vars}(Next)
\]

This formula \( SpecP \) implies that infinitely many \( Next \) steps must occur, so a behavior can’t halt. The weak fairness conjunct therefore requires that \( SpecP \) not set \( p' \) to \( \infty \). This is weird. The liveness property \( WF_{vars}(Next) \) doesn’t just require that something must eventually happen; it also prevents something (setting \( p \) to \( \infty \)) from ever happening. The technical term for this weirdness is that the formula \( SpecP \) is not machine closed [2], which means that its liveness property affects safety as well as liveness.

Non-machine closed specs should never be used to describe how a system works. You can’t understand how to implement a system if the next-state action doesn’t describe what it can and cannot do next. In extremely rare cases, a non-machine closed high-level spec is the best way to describe what a system should
do, rather than how it should do it. You are very unlikely to encounter such a situation in practice.

While the non-machine closed spec we get by adding a prophecy variable to a spec with liveness can be weird, this weirdness causes no problem. We don’t have to implement SpecP. We use it only to check the correctness of Spec; and the presence of the liveness property makes no difference in what we do. With our modified SendInt specifications, we can check that SpecP implies $SI2!Spec$ exactly as we did before.

5 Stuttering Variables

5.1 Adding Stuttering Steps to a Simple Action

Suppose Spec$_1$ is a specification of a (24-hour) clock that displays only the hour—a specification we can write as

\[(5.1) \quad Spec_1 \overset{\triangle}{=} (h = 0) \land \Box [h' = (h + 1) \% 24]_h\]

Let Spec$_2$ be this specification of an hour-minute clock:

\[(5.2) \quad Spec_2 \overset{\triangle}{=} (h = 0) \land (m = 0) \land \Box [m' = (m + 1) \% 60] \land h' = \text{if } m' = 0 \text{ then } (h + 1) \% 24 \text{ else } h ]_{\langle h, m \rangle}\]

If we ignore the variable $m$ in Spec$_2$, then we get a clock that displays only the hour. Thus, $\exists m : Spec_2$ should be equivalent to Spec$_1$. (The 59 steps each hour that change only $m$ are stuttering steps that are allowed by Spec$_1$.) It’s easy to see that Spec$_2$ implies Spec$_1$, so $\exists m : Spec_2$ implies Spec$_1$. There is no refinement mapping with which we can prove that Spec$_1$ implies $\exists m : Spec_2$. To construct the necessary refinement mapping we need to add an auxiliary variable $s$ to Spec$_1$ to obtain a specification Spec$_1^s$ that adds 59 steps that change only $s$ to each step that increments $h$. Such an auxiliary variable is called a stuttering variable because it changes Spec$_1$ only by requiring it to add steps that leave the variable $h$ of Spec$_1$ unchanged.

To add such a variable $s$ to a specification Spec to form Spec$^s$, we let the next-state action of Spec$^s$ take “normal” steps that satisfy the next-state action of Spec when $s$ equals $\top$ (usually read “top”), which is some value that is not a positive integer. The value of $s$ in the initial state equals $\top$. When $s$ is set to a positive integer, the specification Spec$^s$ allows only stuttering steps that decrement $s$, leaving the variables of Spec unchanged. When $s$ counts down to zero, it is set equal to $\top$ again. We add these stuttering steps before and/or after steps of some particular subaction of the next-state action. Here, we assume that this is a “simple” subaction, meaning that its context is empty.
Suppose we want the specification $Spec^s$ to add stuttering steps to each step of a particular subaction. We replace each subaction $A$ by the action $A^s$ defined as follows. For each $A$ other than that particular subaction, we define $A^s$ by:

\[(5.3) \quad A^s \triangleq (s = \top) \land A \land (s' = s)\]

To add $initVal$ stuttering steps after a step of an action $A$, for a positive integer $initVal$ (whose value may depend on the variables of $Spec$), we define

\[
A^s \triangleq \begin{cases} 
\text{IF } s = \top \\
\quad \text{THEN } A \land (s' = initVal) \\
\quad \text{ELSE } \land \text{vars'} = \text{vars} \\
\quad \land s' = \text{IF } s = 1 \text{ THEN } \top \text{ ELSE } s - 1
\end{cases}
\]

We can generalize this by replacing the set of natural numbers with an arbitrary set $\Sigma$ having a well-founded partial order $\prec$ with smallest element $\bot$ (read “bottom”)\(^6\), letting $initVal$ be an arbitrary element of $\Sigma$, replacing $s = 1$ with $s = \bot$, and replacing $s - 1$ by $decr(s)$ for some operator $decr$ such that $decr(s) \prec s$ for all $s \in \Sigma$. The generalization is:

\[(5.4) \quad A^s \triangleq \begin{cases} 
\text{IF } s = \top \\
\quad \text{THEN } A \land (s' = initVal) \\
\quad \text{ELSE } \land \text{vars'} = \text{vars} \\
\quad \land s' = \text{IF } s = \bot \text{ THEN } \top \text{ ELSE } decr(s)
\end{cases}\]

We can add $initVal$ stuttering steps before an $A$ step, rather than after it, as follows. The stuttering steps should only be taken when they can be followed by an $A$ step, which is the case only when an $A$ step is enabled. Remembering that $\text{ENABLED } A$ is the state predicate that is true iff an $A$ step is enabled, the stuttering steps can be added with this definition of $A^s$:

\[(5.5) \quad A^s \triangleq \land \text{ENABLED } A \\
\quad \land \text{IF } s = \bot \text{ THEN } A \land (s' = \top) \\
\quad \text{ELSE } \land \text{vars'} = \text{vars} \\
\quad \land s' = \text{IF } s = \top \text{ THEN } initVal \text{ ELSE } decr(s)\]

We could generalize (5.4) and (5.5) to allow putting stuttering steps both before and after an $A$ step. We won’t bother to do this because it is probably seldom needed, and it wouldn’t be significantly simpler than adding two separate stuttering variables.

\(^6\) This means that $\bot \prec \sigma$ for all $\sigma \in \Sigma$, and any decreasing chain $\sigma_1 \succ \sigma_2 \succ \ldots$ of elements in $\Sigma$ must be finite.
5.2 Adding Stuttering Steps to Multiple Actions

To generalize what we did in the preceding section, we first consider how to add stuttering steps before or after an action $A$ that may have a nonempty context. We assume that the set $\Sigma$, its $\perp$ element, and the operator $decr$ do not depend on the values of the context variables. (This should be true in most real examples.) However, $initVal$ may depend on them. We therefore must make sure that $initVal$ is evaluated with the values of the context variables for which the action $A$ is “executed”. This is no problem for stuttering steps added after the $A$ step, where $A_s$ is defined by (5.4). (Since the stuttering steps do nothing but decrement the value of $s$, it makes no difference for which value of the context variables $A_s$ is “executed” when $s \neq \top$.) However, it is a problem for stuttering steps added before an $A$ step, where $A_s$ is defined by (5.5).

To solve the problem for $A_s$ defined by (5.5), we let the non-$\top$ values of $s$ be records with a $val$ component that equals the value of $s$ described in (5.5), and a $ctxt$ component that equals the tuple of values of the context when $initVal$ is evaluated. In other words, $ctxt$ is set by the $\text{else}$ clause in the second conjunct of (5.5). The condition that $s.ctxt$ equals the values of the context variables is added as a conjunct to the $\text{then}$ clause to make sure that $A$ is executed only in that context. The precise definition is given below.

We often need to add stuttering steps before or after more than one subaction of the next-state action. We could do that by adding separate stuttering variables, or we could introduce a stuttering array variable. However, because the stuttering steps we add to an action all occur immediately before or after that subaction, we can add them all with the same stuttering variable $s$. Stuttering steps can be added to each such subaction with its own set $\Sigma$ and hence its own values of $initVal$ and $\perp$ and of the operator $decr$. We just let the value of $s$ indicate the action to which the stuttering steps are being added. We do this by adding to the non-$\top$ values of $s$ an additional $id$ component that identifies the action for which the stuttering steps are being added. The component is set when $s$ is first set to a non-$\top$ value, and execution of the new subaction $A_s$ is enabled when $s \neq \top$ only if $s.id$ equals the identifier of $A$.

We write the three definitions (5.3), (5.4), and (5.5) in TLA$^+$ using the three operators $\text{NoStutter}$, $\text{PostStutter}$, and $\text{PreStutter}$, respectively, shown in Figure 15. The module should be instantiated by substituting the new stuttering variable for $s$ and the tuple of variables of the original specification for $vars$. It defines $\top$, which is written $top$, to be a value that is different from the values assigned to $s$ by $\text{PostStutter}$ and $\text{PreStutter}$ actions (and that TLC knows is different from those values). The other values in (5.4) and (5.5) are provided by the following arguments to $\text{PostStutter}$ and $\text{PreStutter}$.

- $A$  The action to which stuttering steps are being added.
- $id$ An identifier to distinguish that action $A$ from other actions to which stuttering steps are added. We like to let it be the name of
\[ top \triangleq [\text{top} \mapsto \text{"top"}] \]

VARIABLES \( s, \) \( vars \)

\[ \text{NoStutter}(A) \triangleq (s = \text{top}) \land A \land (s' = s) \]

\[ \text{PostStutter}(A, \text{actionId}, \text{context}, \text{bot}, \text{initVal}, \text{decr}(\_)) \triangleq \]

\[
\text{IF } s = \text{top} \text{ THEN } \land A
\]
\[
\land s' = [\text{id} \mapsto \text{actionId}, \text{ctxt} \mapsto \text{context}, \text{val} \mapsto \text{initVal}]
\]

\[
\text{ELSE } \land s.\text{id} = \text{actionId}
\land \text{UNCHANGED vars}
\land s' = \text{IF } s.\text{val} = \text{bot} \text{ THEN } \text{top}
\]
\[
\text{ELSE } [s \text{ EXCEPT } !.\text{val} = \text{decr}(s.\text{val})]
\]

\[ \text{PreStutter}(A, \text{enabled}, \text{actionId}, \text{context}, \text{bot}, \text{initVal}, \text{decr}(\_)) \triangleq \]

\[
\text{IF } s = \text{top} \text{ THEN } \land \text{enabled}
\land \text{UNCHANGED vars}
\land s' = [\text{id} \mapsto \text{actionId}, \text{ctxt} \mapsto \text{context}, \text{val} \mapsto \text{initVal}]
\]

\[
\text{ELSE } \land s.\text{id} = \text{actionId}
\land \text{IF } s.\text{val} = \text{bot} \text{ THEN } \land s.\text{ctxt} = \text{context}
\land A
\land s' = \text{top}
\]
\[
\text{ELSE } \land \text{UNCHANGED vars}
\land s' = [s \text{ EXCEPT } !.\text{val} = \text{decr}(s.\text{val})]
\]

Figure 15: The beginning of the \textit{Stuttering} module.

the action (which is a string).

\textit{bot} \quad \text{The } \bot \text{ (smallest) element of } \Sigma.

\textit{initVal} \quad \text{The value we have been calling by that name.}

\textit{decr} \quad \text{The operator we have been calling by that name. It must take a single argument.}

\textit{enabled} \quad \text{A formula that should be equivalent to } \text{ENABLED } A. \text{ We can often find such a formula that TLC can evaluate much more efficiently than } \text{ENABLED } A. \text{ You can use TLC to check that } \text{enabled} \text{ is equivalent to } \text{ENABLED } A \text{ by checking that } \text{enabled} \equiv \text{ENABLED } A \text{ is an invariant of the original specification.}

\textit{context} \quad \text{The tuple of context identifiers of } A. \text{ (You can use } i \text{ instead of a 1-tuple } \langle i \rangle.) \text{ The context argument is used in the PostStutter}
\[ \text{MayPostStutter}(A, \text{actionId}, \text{context}, \text{bot}, \text{initVal}, \text{decr}(\_)) \triangleq \]

\[
\begin{align*}
\text{IF } s = \text{top} \text{ THEN } & \land A \\
& \land s' = \text{IF } \text{initVal} = \text{bot} \\
& \text{THEN } s \\
& \text{ELSE } [id \mapsto \text{actionId}, \text{ctxt} \mapsto \text{context}, \\
& \hspace{2cm} \text{val} \mapsto \text{initVal}] \\
\text{ELSE } & \land s.\text{id} = \text{actionId} \\
& \land \text{UNCHANGED vars} \\
& \land s' = \text{IF } \text{decr}(s.\text{val}) = \text{bot} \\
& \text{THEN } \text{top} \\
& \text{ELSE } [s \text{ EXCEPT } !.\text{val} = \text{decr}(s.\text{val})]
\end{align*}
\]

\[ \text{MayPreStutter}(A, \text{enabled}, \text{actionId}, \text{context}, \text{bot}, \text{initVal}, \text{decr}(\_)) \triangleq \]

\[
\begin{align*}
\text{IF } s = \text{top} \text{ THEN } & \land \text{enabled} \\
& \land \text{IF } \text{initVal} = \text{bot} \\
& \text{THEN } A \land (s' = s) \\
& \text{ELSE } \land \text{UNCHANGED vars} \\
& \land s' = [id \mapsto \text{actionId}, \text{ctxt} \mapsto \text{context}, \\
& \hspace{2cm} \text{val} \mapsto \text{decr}(\text{initVal})] \\
\text{ELSE } & \land s.\text{id} = \text{actionId} \\
& \land \text{IF } s.\text{val} = \text{bot} \text{ THEN } \land s.\text{ctxt} = \text{context} \\
& \hspace{2cm} \land A \\
& \hspace{2cm} \land s' = \text{top} \\
& \text{ELSE } \land \text{UNCHANGED vars} \\
& \land s' = [s \text{ EXCEPT } !.\text{val} = \text{decr}(s.\text{val})]
\end{align*}
\]

Figure 16: The end of the Stuttering module.

action only to set the ctxt component of s. This component may be used in defining refinement mappings.

Note that PostStutter and PreStutter add at least one stuttering step, adding exactly one such step if initVal = bot. It is often more convenient to use operators that add one fewer stuttering step. These are the operators MayPostStutter and MayPreStutter defined in the Stuttering module as shown in Figure 16. Unlike the original definitions, the actions they define do not execute any stuttering step when initVal equals bot.

As a simple example, let formula Spec be defined as in Figure 17 to equal the hour clock specification of (5.1). Let us suppose that a module HourMin defines a formula Spec to equal the hour-minute clock specification Spec2 of (5.2). To construct a refinement mapping under which the hour clock specification implements the hour-minute clock specification, we add 59 stuttering steps before...
EXTENDS Integers

VARIABLE h

Init $\triangleq h = 0$

Next $\triangleq h' = (h + 1) \mod 24$

Spec $\triangleq Init \land \square[Next]_h$

Figure 17: The hour clock specification.

each Next step of the hour-clock specification. The obvious way to do that is to let $\Sigma$ be the set 1 . . 59 ordered by $<$, with $\bot$ equal to 1 and initVal equal to 59. However, our refinement mapping becomes simpler if we use the reverse ordering $>$ of 1 . . 59, with $\bot$ equal to 59 and initVal equal to 1. The refinement mapping can then define $\overline{m}$ to equal 0 when $s = \top$ and $s.val$ when $s \neq \top$.

We use the operator PreStutter of the Stuttering module to define $Next^s$. We instantiate that module with vars equal to $h$ and with $s$ equal to the stuttering variable, which we also call $s$. For the arguments of PreStutter, observe that:

- Next is always enabled, so ENABLED Next equals TRUE.

- Since we are adding stuttering steps to only one action, it doesn’t matter what constant we choose for the actionId argument.

- Next, which is the only subaction in the trivial disjunctive representation of Next, has a null context. We can therefore let the context argument be any constant.

We therefore add the following to the end of module Hour.

vars $\triangleq h$

VARIABLE s

INSTANCE Stuttering

InitS $\triangleq Init \land (s = top)$

NextS $\triangleq$ PreStutter(Next, true, “Next”, “”, 59, 1, lambda j : j + 1)

SpecS $\triangleq$ InitS $\land \square[NextS]_{\langle vars, s \rangle}$

HM $\triangleq$ INSTANCE HourMin with $m \leftarrow$ IF $s = top$ THEN 0 ELSE $s.val$

THEOREM SpecS $\Rightarrow$ HM!Spec

TLC can easily check this theorem.
5.3 Correctness of Adding a Stuttering Variable

How do we check that adding a stuttering variable using the operators of the Stuttering module produces a specification $\text{Spec}^s$ such that $\exists s : \text{Spec}^s$ is equivalent to the original specification $\text{Spec}$? The construction ensures that each behavior of $\text{Spec}^s$ is obtained by adding stuttering steps to a behavior of $\text{Spec}$, so $\exists s : \text{Spec}^s$ implies $\text{Spec}$. It can fail to be equivalent to $\text{Spec}$ only if it either adds an infinite sequence of stuttering steps, or if we have used an incorrect $\text{enabled}$ argument for $\text{PreStutter}$. It will be equivalent to $\text{Spec}$ if the following conditions are satisfied for every use of the $\text{PostStutter}$ and $\text{PreStutter}$ operators, with arguments named as above, for some constant set $\Sigma$:

1. For every $\sigma$ in $\Sigma$, the sequence of values $\sigma$, $\text{decr}(\sigma)$, $\text{decr}(\text{decr}(\sigma))$, $\ldots$ is contained in $\Sigma$ and eventually reaches $\text{bot}$.

2. $\text{initVal}$ is in $\Sigma$.

3. $\text{enabled}$ is equivalent to $\text{ENABLED} A$ [for $\text{PreStutter}$ only]

Condition 1 is a condition only on the constants $\Sigma$, $\text{bot}$, and $\text{decr}$. It can be written as $\text{StutterConstantCondition}(\Sigma, \text{bot}, \text{decr})$ using the following definition from the Stuttering module:

$$\text{StutterConstantCondition}(\Sigma, \text{bot}, \text{decr}) \triangleq \text{LET } \text{InverseDecr}(S) \triangleq \{ \text{sig} \in \Sigma \setminus S : \text{decr}(\text{sig}) \in S \}$$
$$R[n \in \text{Nat}] \triangleq \text{IF } n = 0 \text{ THEN } \{ \text{bot} \}$$
$$\text{ELSE LET } T \triangleq R[n - 1]$$
$$\text{IN } T \cup \text{InverseDecr}(T)$$
$$\text{IN } \Sigma = \text{union} \{ R[n] : n \in \text{Nat} \}$$

This condition can be checked by TLC by putting it into an assume statement or else putting it in the Evaluate Constant Expression field of a model’s Model Checking Results page. In either case, the model must replace $\text{Nat}$ by a $0..n$ for a (sufficiently large) integer $n$, and $\Sigma$ must also be replaced with a finite set if it is infinite. The Stuttering module defines $\text{AltStutterConstantCondition}$ to be equivalent to $\text{StutterConstantCondition}$ if $\Sigma$ is finite, and it doesn’t require redefining $\text{Nat}$.

The last two conditions are ones that need only hold for behaviors of $\text{Spec}$. They can be stated formally as follows, where $A$ is a subaction with context $\langle k; K \rangle$:

(5.6) $\text{Spec} \Rightarrow \Box[\forall \langle k; K \rangle : A \Rightarrow (\text{initVal} \in \Sigma)]_{\text{vars}}$

(5.7) $\text{Spec} \Rightarrow \Box(\forall \langle k; K \rangle : \text{enabled} \equiv \text{ENABLED} A)$

TLC can check them in the obvious way.
5.4 Adding Infinite Stuttering

The type of stuttering variable we have been describing adds a finite number of stuttering steps before or after a step of a subaction. There is another type of stuttering variable that adds an infinite number of stuttering steps not associated with an action. You are unlikely ever to have to use one, but we include it for completeness.

Suppose we want to find a refinement mapping under which a spec \( \text{Spec}_1 \) that allows only halting behaviors implements a spec \( \text{Spec}_2 \) that allows behaviors in which externally visible variables stop changing, but internal variables keep changing forever. We obviously can’t do that, because if all the variables of \( \text{Spec}_1 \) stop changing, then no expression defined in terms of those variables can keep changing forever. None of the methods we have described thus far for adding an auxiliary variable \( a \) to \( \text{Spec}_1 \) can help us, because they all have the property that if every behavior allowed by \( \text{Spec}_1 \) halts, then so does every behavior allowed by \( \text{Spec}_1^a \).

It’s hard to devise a practical example in which this problem would arise. One possibility is for \( \text{Spec}_2 \) to have a server perform internal actions looking for user input that may never arrive, while in \( \text{Spec}_1 \) the server just waits for input. Although unlikely to arise, the problem is easy to solve, so we briefly sketch a solution.

The solution is to add a stuttering variable that is required to stutter forever. Let \( \text{Spec} \) equal \( \text{Init} \wedge \Box[\text{Next}]_{\text{vars}} \) and let \( \text{UC} \) be the stuttering action UNCHANGED \( \text{vars} \). Since \( [\text{Next}]_{\text{vars}} \) equals \( \text{Next} \lor \text{UC} \), we can write \( \text{Spec} \) as \( \text{Init} \wedge \Box[\text{Next} \lor \text{UC}]_{\text{vars}} \). We can therefore add the subaction \( \text{UC} \) to any disjunctive representation of \( \text{Next} \). We add a history variable \( s \) as described in Section 3, using a disjunctive representation containing the subaction \( \text{UC} \). This defines an action \( A^s \) for every subaction \( A \) and produces the specification \( \text{Init}^s \wedge \Box[\text{Next}^s]_{\langle \text{vars}, s \rangle} \). (We choose \( \text{UC}^s \) so it implies \( s' \neq s \).) We then define \( \text{Spec}^s \) to equal

\[
\text{Init}^s \wedge \Box[\text{Next}^s]_{\langle \text{vars}, s \rangle} \wedge \text{WF}_{\langle \text{vars}, s \rangle}(\text{UC}^s)
\]

Since enabled \( \text{UC}^s \) equals \text{true}, the fairness requirement \( \text{WF}_{\langle \text{vars}, s \rangle}(\text{UC}^s) \) implies that infinitely many \( \text{UC}^s \) steps occur. These are steps that leave the variables in \( \text{vars} \) unchanged, changing only \( s \). Since we have added \( s \) as a history variable, \( \exists s : \text{Init}^s \wedge \Box[\text{Next}^s]_{\langle \text{vars}, s \rangle} \) is equivalent to \( \text{Spec} \). Since any TLA\(^+\) spec allows stuttering steps, this implies that \( \exists s : \text{Spec}^s \) is equivalent to \( \text{Spec} \).

5.5 Liveness

Liveness poses no problem when adding a stuttering variable. As with other auxiliary variables, we obtain \( \text{Spec}^s \) by adding the stuttering variable to the safety part of \( \text{Spec} \) and then conjoining to it the liveness conjunct of \( \text{Spec} \). (This
Although $Spec^s$ may have an unusual form, it isn’t weird. If $Spec$ is machine closed then $Spec^s$ is also machine closed. However, putting it into a standard form with fairness conditions only on subactions of $Next^s$ is not as simple as it is for history variables.

6 The Snapshot Problem

We now consider an example of using auxiliary variables to show that an algorithm satisfies its specification. Our example is based on an algorithm of Afek et al. [3]. Their algorithm implements what they call a single-writer atomic snapshot memory, which will call simply a snapshot object. Their algorithm implements a snapshot object using an unbounded amount of storage. They also present a second algorithm that uses a bounded amount of storage and implements a more general type of object, but we restrict ourselves to their first, simpler algorithm. Moreover, we consider only a simplified version of this simpler algorithm; their algorithm can be checked by adding the same auxiliary variables used for the simplified version.

6.1 Linearizability

A snapshot algorithm is used to implement an atomic read of an array of memory registers, each of which can be written by a different process. Its specification is a special case of a linearizable specification of a data object—a concept introduced by Herlihy and Wing [4].

A data object, also called a state machine, executes commands from user processes. It is described by an initial state of the object and an operator $Apply$, where $Apply(i, cmd, st)$ describes the output and new state of the object that results from process $i$ executing command $cmd$ when the object has state $st$. It is specified formally by these declared constants:

$$\text{CONSTANTS} \quad Procs, \ Commands(_), \ Outputs(_), \ InitObj, \ Apply(_, _, _)$$

They have the following meanings:

$Procs$ The set of processes.

$Commands(i)$ The set of commands that process $i$ can issue.

$Outputs(i)$ The set of outputs the commands issued by process $i$ can produce.

$InitObj$ The initial state of the object.
Apply\((i, \text{cmd}, \text{st})\) A record with output and newState fields describing the result of process \(i\) executing command \(\text{cmd}\) when the object is in state \(\text{st}\).

A linearizable implementation of the data object is one in which the state of the object is internal, the only externally visible actions being the issuing of the command and the return of its output. More precisely, a process \(i\) executes a command \(\text{cmd}\) with a BeginOp\((i, \text{cmd})\) step, followed by a DoOp\((i)\) step that modifies the state of the object, followed by an EndOp\((i)\) step. The BeginOp and EndOp steps are externally visible, meaning that they modify externally visible variables (and perhaps internal variables), while the DoOp step modifies only internal variables—including an internal variable describing the state of the object.

To simplify the specification, we assume that the sets of commands and of outputs are disjoint. We can then use a single externally visible variable interface, letting BeginOp\((i, \text{cmd})\) set \(\text{interface}[i]\) to the command \(\text{cmd}\) and letting EndOp\((i)\) set it to the command’s output. We also introduce an internal variable istate to hold the internal state of the processes—needed to remember, while a process is executing a command, whether or not it has performed the DoOp step and, if it has, what output was produced. We do this by letting BeginOp\((i, \text{cmd})\) set \(\text{istate}[i]\) to \(\text{cmd}\), and letting DoOp\((i)\) set it to the command’s output. Here is the definition of the next-state action and its sub-actions.

\[
\begin{align*}
\text{BeginOp}(i, \text{cmd}) & \triangleq \land \text{interface}[i] \in \text{Outputs}(i) \\
& \land \text{interface}' = [\text{interface \ except \ !}[i] = \text{cmd}] \\
& \land \text{istate}' = [\text{istate \ except \ !}[i] = \text{cmd}] \\
& \land \text{object}' = \text{object}
\end{align*}
\]

\[
\begin{align*}
\text{DoOp}(i) & \triangleq \land \text{interface}[i] \in \text{Commands}(i) \\
& \land \text{istate}[i] = \text{interface}[i] \\
& \land \text{LET result} \triangleq \text{Apply}(i, \text{interface}[i], \text{object}) \\
& \quad \land \text{object}' = \text{result}.\text{newState} \\
& \quad \land \text{istate}' = [\text{istate \ except \ !}[i] = \text{result}.\text{output}] \\
& \land \text{interface}' = \text{interface}
\end{align*}
\]

\[
\begin{align*}
\text{EndOp}(i) & \triangleq \land \text{interface}[i] \in \text{Commands}(i) \\
& \land \text{istate}[i] \in \text{Outputs}(i) \\
& \land \text{interface}' = [\text{interface \ except \ !}[i] = \text{istate}[i]] \\
& \land \text{UNCHANGED \ <object, istate>}
\end{align*}
\]

\[
\text{Next} \triangleq \exists i \in \text{Procs} : \lor \exists \text{cmd} \in \text{Commands}(i) : \text{BeginOp}(i, \text{cmd}) \\
\lor \text{DoOp}(i) \\
\lor \text{EndOp}(i)
\]

Initially, interface\([i]\) and istate\([i]\) equal some output, for each \(i\). We let that
equal $InitOutput(i)$ for some constant operator $InitOutput$. We also add a fairness requirement to imply that any command that has begun (with a $BeginOp$ step) eventually completes (with an $EndOp$ step). The complete specification (with the action definitions above elided) is in module $Linearizability$, shown in Figure 18. Any particular linearizable object can be specified by instantiating the module with the appropriate constants. The module includes an assumption named $LinearAssumps$, to check that the instantiated constants satisfy the properties they should for the module to specify a linearizable object. To state all those properties, the specification introduces a constant $ObjValues$ to describe the set of all possible states of the object. This set could be defined to equal the following rather complicated expression. Trying to understand it provides a good lesson in set theory.
LET $ApplyProcTo(i, S) \triangleq \{ Apply(i, cmd, x).newState : x \in S, cmd \in Commands(i) \}$

$ApplyTo(S) \triangleq \text{UNION} \{ ApplyProcTo(i, S) : i \in \text{Procs} \}$

$ApplyITimes[i \in \text{Nat}] \triangleq$

\[
\begin{align*}
\text{IF } i = 0 \text{ THEN } & \{ \text{InitObj} \} \\
\text{ELSE } & \text{ApplyTo}(ApplyITimes[i - 1]) \\
\text{IN } & \text{UNION} \{ ApplyITimes[i] : i \in \text{Nat} \}
\end{align*}
\]

6.2 The Linearizable Snapshot Specification

By a snapshot object, we mean what Afek et al. [3] called an **atomic snapshot memory**. In a snapshot object, the processes are either readers or writers. Reader and writer should be thought of as roles; the same physical process can act as both a reader and a writer. A snapshot object is an array of registers, one per writer. A write operation writes a value to the writer’s register and produces as output some fixed value that is not a possible register value. A read operation has a single command that produces the object’s state (an array of register values) as output and leaves that state unchanged.

The specification declares four constants: the sets $\text{Readers}$ and $\text{Writers}$ of reader and writer processes; the set $\text{RegVals}$ of possible register values; and a value $\text{InitRegVal}$ in $\text{RegVals}$ that is the initial value of a register. We call the snapshot object a **memory** and use different names for some of the parameters of the $\text{Linearizability}$ module, including $\text{MemVals}$ and $\text{InitMem}$ for $\text{ObjValues}$ and $\text{InitObj}$. We define $\text{NotMemVal}$ be the single reader command and $\text{NotRegVal}$ to be the single write command output. The complete specification is in module $\text{LinearSnapshot}$ of Figure 19. (The $\text{assume}$ is added at the end of the module so TLC will check that the assumption $\text{LinearAssumps}$ of module $\text{Linearizability}$ is true under the instantiation.)

6.3 The Simplified Afek et al. Snapshot Algorithm

The snapshot algorithm of Afek et al. uses an internal variable $imem$ whose value is an array with $imem[i]$ a pair consisting of the value of the $i^{\text{th}}$ register and an integer whose value is the number of times the register has been written. It assumes that the entire pair can be read and written atomically.

A write operation writes the register value $cmd$ in the obvious way, the $\text{DoOp}(i)$ action setting $imem[i]$ to $\langle cmd, imem[i][2] + 1 \rangle$.

A read operation first performs the following $\text{scan}$ procedure:

It reads all the elements $imem[i]$ once, in any order. It then reads them a second time, again in any order. If it reads the same values both times for all $i$, it outputs the array of register values it read.
\[
\text{MODULE LinearSnapshot}
\]

\text{CONSTANTS Readers, Writers, RegVals, InitRegVal}

\text{ASSUME} \quad \wedge \text{Readers} \cap \text{Writers} = \{\}
\quad \wedge \text{InitRegVal} \in \text{RegVals}

\text{Procs} \triangleq \text{Readers} \cup \text{Writers}

\text{MemVals} \triangleq [\text{Writers} \to \text{RegVals}]
\text{InitMem} \triangleq [i \in \text{Writers} \mapsto \text{InitRegVal}]

\text{NotMemVal} \triangleq \text{CHOOSE } v : v \notin \text{MemVals}
\text{NotRegVal} \triangleq \text{CHOOSE } v : v \notin \text{RegVals}

\text{Commands}(i) \triangleq \text{IF } i \in \text{Readers} \text{ THEN } \{\text{NotMemVal}\}
\quad \text{ELSE } \text{RegVals}

\text{Outputs}(i) \triangleq \text{IF } i \in \text{Readers} \text{ THEN } \text{MemVals}
\quad \text{ELSE } \{\text{NotRegVal}\}

\text{InitOutput}(i) \triangleq \text{IF } i \in \text{Readers} \text{ THEN } \text{InitMem} \text{ ELSE } \text{NotRegVal}

\text{Apply}(i, \text{cmd}, \text{obj}) \triangleq \text{IF } i \in \text{Readers}
\quad \text{THEN } [\text{newState} \mapsto \text{obj}, \text{output} \mapsto \text{obj}]
\quad \text{ELSE } [\text{newState} \mapsto [\text{obj EXCEPT } ![i] = \text{cmd}],
\quad \text{output} \mapsto \text{NotRegVal}]

\text{VARIABLES mem, interface, istate}

\text{INSTANCE Linearizability WITH ObjValues} \leftarrow \text{MemVals}, \text{InitObj} \leftarrow \text{InitMem},
\quad \text{object} \leftarrow \text{mem}

\text{ASSUME LinearAssumps}

Figure 19: Module LinearSnapshot.

If the values obtained for each element \text{imem}[i] by the two reads are not all the same, so the scan procedure does not produce an output, then the procedure is repeated. The scan procedure is repeated again and again until it produces an output.

The actual algorithm has an alternative method of producing an output that can be used when it has read three different values for \text{imem}[i], for some writer \text{w}. By using this method, termination of the read is assured. However, for simplicity, we use an algorithm that keeps performing the scan procedure until it succeeds in producing an output. Thus, a read need never terminate, so the algorithm does not satisfy the liveness requirement of a snapshot algorithm—
namely, it does not satisfy the weak fairness requirement of the \textit{DoOp}(i) action for a reader \textit{i}. However, it does satisfy the safety requirement. The correctness of the complete algorithm (including liveness) can be verified by essentially the same method used for our simplified version; but the complete algorithm is more complicated, so the refinement mapping is more complicated and model checking takes longer. We therefore consider only the simplified algorithm.

To specify the algorithm in TLA$^+$, we declare the same constants \textit{Readers}, \textit{Writers}, \textit{RegVals}, and \textit{InitRegVal} and make the same definitions of \textit{MemVals}, \textit{InitMem}, \textit{NotMemVal}, and \textit{NotRegVal} as in module \textit{LinearSnapshot} above. We also define:

\[
\begin{align*}
I\text{RegVals} & \triangleq \text{RegVals} \times \text{Nat} \\
I\text{MemVals} & \triangleq [\text{Writers} \to I\text{RegVals}] \\
\text{InitIMem} & \triangleq [i \in \text{Writers} \mapsto \langle \text{InitRegVal}, 0 \rangle]
\end{align*}
\]

We declare five variables, with the following meanings:

\textit{interface}: The same as in \textit{LinearSnapshot}.

\textit{imem}: Like \textit{mem} in \textit{LinearSnapshot}, except \textit{imem}[i] is an ordered pair in \textit{RegVals} \times \textit{Nat}, the first component representing \textit{mem}[i] and the second the number of times \textit{mem}[i] has been written. The initial value of \textit{imem} is initially \textit{InitIMem}.

\textit{wrNum}: A function with domain \textit{Writers}, where \textit{wrNum}[i] is the number of \textit{BeginWr}(i) steps that have been taken.

\textit{rdVal1}, \textit{rdVal2}: They are functions such that \textit{rdVal1}[i] and \textit{rdVal2}[i] describe the values read so far by reader \textit{i} in the two reads of the \textit{scan} procedure. Both \textit{rdVal1}[i] and \textit{rdVal2}[i] are functions whose domain is the set of writers \textit{j} for which the first or second read of \textit{imem}[j] has been performed, mapping each such \textit{j} to the value read. They are set initially to the empty function (the function with empty domain), which we write \langle \rangle.

The writer actions are straightforward. Note that because \textit{wrNum}[i] counts the number of \textit{BeginWr}(i, \textit{cmd}) steps and \textit{imem}[i][2] is set to \textit{wrNum}[i] by the \textit{DoWr}(i), the \textit{EndWrite}(i) action should be enabled and \textit{DoWr}(i) disabled when \textit{imem}[i][2] equals \textit{wrNum}[i].

\[
\begin{align*}
\text{BeginWr}(i, \textit{cmd}) & \triangleq \land \text{interface}[i] = \text{NotRegVal} \\
& \land \textit{wrNum}' = [\textit{wrNum} \except ![i] = \textit{wrNum}[i] + 1] \\
& \land \text{interface}' = [\text{interface} \except ![i] = \textit{cmd}] \\
& \land \text{UNCHANGED} \langle \textit{imem}, \textit{rdVal1}, \textit{rdVal2} \rangle \\
\text{DoWr}(i) & \triangleq \land \text{interface}[i] \in \text{RegVals} \\
& \land \textit{imem}[i][2] \neq \textit{wrNum}[i] \\
& \land \textit{imem}' = [\textit{imem} \except ![i] = \langle \text{interface}[i], \textit{wrNum}[i] \rangle]
\end{align*}
\]
\[ \text{EndWr}(i) \triangleq \land \text{interface}[i] \in \text{RegVals} \\
\land \text{imem}[i][2] = \text{wrNum}[i] \\
\land \text{interface}' = [\text{interface} \setminus \text{!}[i] = \text{NotRegVal}] \\
\land \text{UNCHANGED} \langle \text{imem}, \text{wrNum}, \text{rdVal1}, \text{rdVal2} \rangle \]

The \text{BeginRd}(i) action is straightforward.

\[ \text{BeginRd}(i) \triangleq \land \text{interface}[i] \in \text{MemVals} \\
\land \text{interface}' = [\text{interface} \setminus \text{!}[i] = \text{NotMemVal}] \\
\land \text{UNCHANGED} \langle \text{imem}, \text{wrNum}, \text{rdVal1}, \text{rdVal2} \rangle \]

The definitions of the actions that perform the \textit{scan} procedure use the following definition. We define \text{AddToFcn}(f, x, v) to be the function \( g \) obtained from the function \( f \) by adding \( x \) to its domain and defining \( g[x] \) to equal \( v \). Using operators defined in the \textit{TLC} module, it can be defined to equal \( f \circ \circ \circ (x : > v) \). However, it’s easy enough to define it directly as:

\[ \text{AddToFcn}(f, x, v) \triangleq [y \in (\text{DOMAIN}\ f) \cup \{x\} \mapsto \text{IF } y = x \text{ THEN } v \text{ ELSE } f[y]] \]

Using \text{AddToFcn}, we define the \text{Rd1} action that performs the scan’s first read of \text{imem} and the \text{Rd2} action that performs its second read.

\[ \text{Rd1}(i) \triangleq \land \text{interface}[i] = \text{NotMemVal} \\
\land \exists j \in \text{Writers} \setminus \text{DOMAIN } \text{rdVal1}[i] : \\
\text{rdVal1}' = [\text{rdVal1} \setminus \text{!}[i] = \text{AddToFcn(\text{rdVal1}[i], j, \text{imem}[j])}] \\
\land \text{UNCHANGED} \langle \text{interface}, \text{imem}, \text{wrNum}, \text{rdVal2} \rangle \]

\[ \text{Rd2}(i) \triangleq \land \text{interface}[i] = \text{NotMemVal} \\
\land \text{DOMAIN } \text{rdVal1}[i] = \text{Writers} \\
\land \exists j \in \text{Writers} \setminus \text{DOMAIN } \text{rdVal2}[i] : \\
\text{rdVal2}' = [\text{rdVal2} \setminus \text{!}[i] = \text{AddToFcn(\text{rdVal2}[i], j, \text{imem}[j])}] \\
\land \text{UNCHANGED} \langle \text{interface}, \text{imem}, \text{wrNum}, \text{rdVal1} \rangle \]

Finally, we define \text{TryEndRd}(i) to be an action that is enabled when the reader’s scan operation has completed. It compares the values read by the two sets of reads and, if they are equal, it performs the \text{EndOp} for the read. Otherwise, it enables the next scan to begin.

\[ \text{TryEndRd}(i) \triangleq \land \text{interface}[i] = \text{NotMemVal} \\
\land \text{DOMAIN } \text{rdVal1}[i] = \text{Writers} \\
\land \text{DOMAIN } \text{rdVal2}[i] = \text{Writers} \]
∧ IF \( \text{rdVal}1[i] = \text{rdVal}2[i] \) THEN \( \text{interface}' = \)
\[\text{interface EXCEPT } ![i] = [j \in \text{Writers} \mapsto \text{rdVal}1[i][j][1]]\]
ELSE \( \text{interface}' = \text{interface} \)
\( \land \text{rdVal}1' = [\text{rdVal}1 \text{EXCEPT } ![i] = \langle \rangle] \)
\( \land \text{rdVal}2' = [\text{rdVal}2 \text{EXCEPT } ![i] = \langle \rangle] \)
\( \land \text{UNCHANGED } \langle \text{imem, wrNum} \rangle \)

The complete specification is in module \textit{AfekSimplified}, shown in Figure 20 with the action definitions above elided.

6.4 Another Snapshot Specification

The algorithm in module \textit{AfekSimplified} satisfies the safety specification in \textit{LinearSnapshot}, but we now show that it does not implement that safety specification under any refinement mapping. Let \( \text{Spec}_A \) be the algorithm’s specification and let \( \text{SSpec}_L \) be the specification \( \text{SafeSpec} \) of \textit{LinearSnapshot}. We assume there is a refinement mapping \( \text{mem} \leftarrow \overline{\text{mem}} \) and \( \text{istate} \leftarrow \overline{\text{istate}} \) under which \( \text{Spec}_A \) implements \( \text{SSpec}_L \) and obtain a contradiction. Let \( \overline{F} \) be the formula obtained from a formula \( F \) of module \textit{LinearSnapshot} by replacing \( \text{mem} \) with \( \overline{\text{mem}} \) and \( \text{istate} \) with \( \overline{\text{istate}} \). Consider a behavior satisfying \( \text{Spec}_A \) that begins with the following three sequences of steps.

1. Reader \( i \) does a \( \text{BeginRd}(i) \) step, completes its first scan (so \( \text{DOMAIN} \ \text{rdVal}1[i] \) equals \( \text{Writers} \)) and begins its second scan by reading \( \text{imem}[j] = \langle v_1, 0 \rangle \) for some writer \( j \) and \( v_1 \) in \( \text{RegVals} \) (so \( \text{DOMAIN} \ \text{rdVal}2[i] \) equals \( \{j\} \) and \( \text{rdVal}2[i][j] \) equals \( \langle v_1, 0 \rangle \)).

2. Writer \( j \) then does a complete write operation, writing a new value \( v_2 \) different from \( v_1 \).

3. Reader \( i \) completes its second scan, executes its \( \text{TryEndRd}(i) \) action, finding \( \text{rdVal}2[i] \) equal to \( \text{rdVal}1[i] \), and completing the read operation by setting \( \text{interface}[i] \) to a value \( M \) with \( M[j] = v_1 \).

The behavior satisfies \( \overline{\text{SSpec}_L} \), so this sequence of actions must start with a \( \overline{\text{BeginRd}}(i) \) step, contain a \( \overline{\text{DoRd}}(i) \) step, and end with an \( \overline{\text{EndRd}}(i) \) step. The reader has not determined the value to be output by the read command until it has finished its second scan, so the \( \overline{\text{DoRd}}(i) \) step must occur in sequence 3. The three steps of the write of \( v_2 \) by writer \( j \) occur in sequence 2, so the \( \overline{\text{DoWr}}(j) \) step for that write must occur in that sequence, therefore preceding the \( \overline{\text{DoRd}}(i) \) step. Hence, the \textit{LinearSnapshot} spec implies that the \( \overline{\text{DoRd}}(i) \) step must set \( \overline{\text{istate}}[i][j] \) to \( v_2 \). However in the last step of 3, the reader sets the value of \( \text{interface}[i][j] \) to \( v_1 \), which implies that the \( \overline{\text{DoRd}}(i) \) step set \( \overline{\text{istate}}[i][j] \) to \( v_1 \).
MODULE AfekSimplified

EXTENDS Integers

CONSTANTS Readers, Writers, RegVals, InitRegVal

\[
\begin{align*}
\text{MemVals} & \triangleq [\text{Writers} \rightarrow \text{RegVals}] \\
\text{InitMem} & \triangleq [i \in \text{Writers} \mapsto \text{InitRegVal}] \\
\text{NotMemVal} & \triangleq \text{choose } v : v \notin \text{MemVals} \\
\text{NotRegVal} & \triangleq \text{choose } v : v \notin \text{RegVals} \\
\end{align*}
\]

\[
\begin{align*}
\text{IRegVals} & \triangleq \text{RegVals} \times \text{Nat} \\
\text{IMemVals} & \triangleq [\text{Writers} \rightarrow \text{IRegVals}] \\
\text{InitIMem} & \triangleq [i \in \text{Writers} \mapsto \langle \text{InitRegVal}, 0 \rangle] \\
\end{align*}
\]

VARIABLES imem, interface, wrNum, rdVal1, rdVal2
\[
\text{vars} \triangleq \langle \text{imem}, \text{interface}, \text{wrNum}, \text{rdVal1}, \text{rdVal2} \rangle
\]

\[
\begin{align*}
\text{Init} & \triangleq \land \text{imem} = \text{InitIMem} \\
& \land \text{interface} = [i \in \text{Readers} \cup \text{Writers} \mapsto \\
& \quad \text{if } i \in \text{Readers} \text{ then } \text{InitMem} \text{ else } \text{NotRegVal}] \\
& \land \text{wrNum} = [i \in \text{Writers} \mapsto 0] \\
& \land \text{rdVal1} = [i \in \text{Readers} \mapsto \langle \rangle] \\
& \land \text{rdVal2} = [i \in \text{Readers} \mapsto \langle \rangle] \\
\end{align*}
\]

\[
\begin{align*}
\text{BeginWr}(i, \text{cmd}) & \triangleq \ldots \\
\text{DoWr}(i) & \triangleq \ldots \\
\text{EndWr}(i) & \triangleq \ldots \\
\text{BeginRd}(i) & \triangleq \ldots \\
\text{AddToFcn}(f, x, v) & \triangleq \ldots \\
\text{Rd1}(i) & \triangleq \ldots \\
\text{Rd2}(i) & \triangleq \ldots \\
\text{TryEndRd}(i) & \triangleq \ldots \\
\end{align*}
\]

\[
\begin{align*}
\text{Next} & \triangleq \lor \exists i \in \text{Readers} : \text{BeginRd}(i) \lor \text{Rd1}(i) \lor \text{Rd2}(i) \lor \text{TryEndRd}(i) \\
& \lor \exists i \in \text{Writers} : \lor \exists \text{cmd} \in \text{RegVals} : \text{BeginWr}(i, \text{cmd}) \\
& \lor \text{DoWr}(i) \lor \text{EndWr}(i) \\
\end{align*}
\]

\[
\begin{align*}
\text{Spec} & \triangleq \text{Init} \land \Box [\text{Next}]_{\text{vars}}
\end{align*}
\]

Figure 20: Module AfekSimplified
Since $v_1 \neq v_2$, this is a contradiction, showing that the refinement mapping cannot exist.

This behavior of $Spec_A$ is allowed by $\exists \ mem, istate : SSpec_L$, since we can choose values of $\overset{\text{DoRd}(i)}{mem}$ and $\overset{\text{istate}}{istate}$ for which $SSpec_L$ is satisfied—namely, values for which the $\overset{\text{DoRd}(i)}{\text{DoRd}}$ step occurs before the $\overset{\text{DoWr}(j)}{\text{DoWr}}$ step. However, choosing those values requires knowing what steps occur after the $\overset{\text{DoWr}(j)}{\text{DoWr}}$ step. The linearizability specification $SSpec_L$ chooses the value returned by a read sooner than it has to. This tells us that to find a refinement mapping that shows $Spec_A$ implements $\exists \ mem, istate : SSpec_L$, we must add a prophecy variable to $Spec_A$.

Instead of adding a prophecy variable to $Spec_A$, we write a new snapshot specification $Spec_{NL}$ that allows the same externally visible behaviors as specification $Spec_L$ of $LinearSnapshot$; and whose safety specification $SSpec_{NL}$ allows the same visible behaviors as $SSpec_L$. However, in $Spec_{NL}$ we make a reader wait as long as possible before choosing its output value. We can then find a refinement mapping to show that $Spec_A$ implements $SSpec_{NL}$ without using a prophecy variable.

We will still need a prophecy variable to show that $SSpec_{NL}$ allows the same externally visible behavior as $SSpec_L$. The advantage of introducing $Spec_{NL}$ is that the specification of what an algorithm is supposed to do is generally much simpler than the algorithm. Prophecy variables are the most complicated kind of auxiliary variables, and it is easier to add one to a high-level specification than to a lower-level algorithm. (This same idea of modifying the high-level specification to avoid adding a prophecy variable to the algorithm can be applied to the queue example of Herlihy and Wing [4].)

Specification $Spec_{NL}$ records in its internal state all values of the memory $\overset{\text{mem}}{mem}$ that a read operation is allowed to return. The $\overset{\text{EndRd}}{\text{EndRd}}$ operation nondeterministically chooses one of those values as its output. Its internal state therefore remembers much more about what happened in the past than a reasonable implementation would. This means that defining a refinement mapping under which an algorithm implements $Spec_{NL}$ will require adding a history variable to the algorithm’s spec. Adding a history variable is much easier than adding a prophecy variable.

We write specification $Spec_{NL}$ (and $SSpec_{NL}$) in module $NewLinearSnapshot$. It has the same declarations of $\overset{\text{Readers}}{Readers}$, $\overset{\text{Writers}}{Writers}$, $\overset{\text{RegVals}}{\text{RegVals}}$, and $\overset{\text{InitRegVal}}{\text{InitRegVal}}$ and the same definitions of $\overset{\text{MemVals}}{\text{MemVals}}$, $\overset{\text{InitMem}}{\text{InitMem}}$, $\overset{\text{NotMemVal}}{\text{NotMemVal}}$, and $\overset{\text{NotRegVal}}{\text{NotRegVal}}$ as in module $LinearSnapshot$. It has the same variables $\overset{\text{interface}}{interface}$ and $\overset{\text{mem}}{\overset{\text{mem}}{\text{mem}}}$ as module $LinearSnapshot$, plus these two internal variables:

\begin{itemize}
\item \textit{wstate} A function with domain $\overset{\text{Writers}}{Writers}$ such that the value $\overset{\text{wstate}[i]}{\text{wstate}[i]}$ is the same as the value of $\overset{\text{istate}[i]}{\text{istate}[i]}$ in $LinearSnapshot$, for each writer $i$.
\item \textit{rstate} A function with domain $\overset{\text{Readers}}{Readers}$ so that, for each reader $i$ currently executing a read operation, $\overset{\text{rstate}[i]}{\text{rstate}[i]}$ is the sequence of values that $\overset{\text{mem}}{\overset{\text{mem}}{\text{mem}}}$ has assumed thus far while the operation has been executing.
\end{itemize}
The first element of $rstate[i]$ is therefore the value $mem$ had when the $BeginRd(i)$ step occurred. The value of $rstate[i]$ is the empty sequence $⟨⟩$ when $i$ is not executing a read operation.

The $BeginWr$ command is essentially the same as in $LinearSnapshot$.

$$BeginWr(i, cmd) \triangleq \land \text{interface}[i] = \text{NotRegVal}$$
$$\land \text{interface}' = [\text{interface} \ \text{EXCEPT} \ ![i] = cmd]$$
$$\land \text{wstate}' = [\text{wstate} \ \text{EXCEPT} \ ![i] = \langle mem \rangle]$$
$$\land \text{UNCHANGED} \langle \text{mem}, \ rstate \rangle$$

The $BeginRd(i)$ command, which sets $rstate[i]$ to a one-element sequence containing the current value of $mem$, is:

$$BeginRd(i) \triangleq \land \text{interface}[i] \in \text{MemVals}$$
$$\land \text{interface}' = [\text{interface} \ \text{EXCEPT} \ ![i] = \text{NotMemVal}]$$
$$\land \text{rstate}' = [\text{rstate} \ \text{EXCEPT} \ ![i] = \langle mem \rangle]$$
$$\land \text{UNCHANGED} \langle \text{mem}, \ rstate \rangle$$

The writer executes a $DoWr$ that is the same as in $LinearSnapshot$, except that it also appends the new value of $mem$ to the end of $rstate[j]$ for every reader $j$ currently executing a read operation.

$$DoWr(i) \triangleq \land \text{interface}[i] \in \text{RegVals}$$
$$\land \text{wstate}[i] = \text{interface}[i]$$
$$\land \text{mem}' = [\text{mem} \ \text{EXCEPT} \ ![i] = \text{interface}[i]]$$
$$\land \text{wstate}' = [\text{wstate} \ \text{EXCEPT} \ ![i] = \text{NotRegVal}]$$
$$\land \text{rstate}' = [j \in \text{Readers} \mapsto$$
$$\text{IF} \ rstate[j] = \langle \rangle$$
$$\text{THEN} \ \langle \rangle$$
$$\text{ELSE} \ \text{Append}(rstate[j], \text{mem}')]$$
$$\land \text{interface}' = \text{interface}$$

A reader $i$ has no internal actions, only the externally visible $BeginRd(i)$ and $EndRd(i)$ actions. Its $EndRd(i)$ action outputs an arbitrarily chosen element of $rstate[i]$.

$$EndRd(i) \triangleq \land \text{interface}[i] = \text{NotMemVal}$$
$$\land \exists j \in 1..\text{Len}(rstate[i]) :$$
$$\text{interface}' = [\text{interface} \ \text{EXCEPT} \ ![i] = \text{rstate}[i][j]]$$
$$\land \text{rstate}' = [\text{rstate} \ \text{EXCEPT} \ ![i] = \langle \rangle]$$
$$\land \text{UNCHANGED} \langle \text{mem}, \ rstate, \ wstate \rangle$$

The writer’s $EndWr$ action is essentially the same as in $LinearSnapshot$.

$$EndWr(i) \triangleq \land \text{interface}[i] \in \text{RegVals}$$
$$\land \text{wstate}[i] = \text{NotRegVal}$$
$$\land \text{interface}' = [\text{interface} \ \text{EXCEPT} \ ![i] = \text{wstate}[i]]$$
$$\land \text{UNCHANGED} \langle \text{mem}, \ rstate, \ wstate \rangle$$

The complete module, minus the action definitions above, is in Figure 21.
MODULE NewLinearSnapshot

EXTENDS Integers, Sequences

CONSTANTS Readers, Writers, RegVals, InitRegVal

ASSUME ∧ Readers ∩ Writers = {}
∧ InitRegVal ∈ RegVals

InitMem ≜ [i ∈ Writers ↦ InitRegVal]
MemVals ≜ [Writers → RegVals]
NotMemVal ≜ \text{choose } v : v ∉ MemVals
NotRegVal ≜ \text{choose } v : v ∉ RegVals

VARIABLES mem, interface, rstate, wstate
vars ≜ ⟨mem, interface, rstate, wstate⟩

Init ≜ ∧ mem = InitMem
∧ interface = [i ∈ Readers ∪ Writers ↦
IF \quad i ∈ Readers \quad THEN \quad InitMem \quad ELSE \quad NotRegVal]
∧ rstate = [i ∈ Readers ↦ ⟨⟩]
∧ wstate = [i ∈ Writers ↦ NotRegVal]

BeginRd(i) ≜ ...
BeginWr(i, cmd) ≜ ...
DoWr(i) ≜ ...
EndRd(i) ≜ ...
EndWr(i) ≜ ...

Next ≜ ∨ \exists i ∈ Readers : BeginRd(i) \lor EndRd(i)
∨ \exists i ∈ Writers : \exists cmd ∈ RegVals : BeginWr(i, cmd)
\lor DoWr(i) \lor EndWr(i)

SafeSpec ≜ Init ∧ □[Next]vars

Fairness ≜ ∧ ∀ i ∈ Readers : WFvars(EndRd(i))
∧ ∀ i ∈ Writers : WFvars(DoWr(i)) \land WFvars(EndWr(i))

Spec ≜ Init ∧ □[Next]vars \land Fairness

Figure 21: Module NewLinearSnapshot
NewLinearSnapshot Implements LinearSnapshot

For compactness, in the following discussion we let:

\[ S_L \triangleq \exists \text{mem, istate} : \text{Spec}_L \]
\[ S_{NL} \triangleq \exists \text{mem, rstate, wstate} : \text{Spec}_{NL} \]

Specifications \( S_L \) and \( S_{NL} \) are equivalent. However, our goal is to prove that \( \text{Spec}_A \) implements \( S_L \), for which it suffices to show that it implements specification \( S_{NL} \) and that \( S_{NL} \) implements \( S_L \). So, we won’t bother showing equivalence of the two specs; we just show here that \( S_{NL} \) implements \( S_L \). We show in Section 6.6 below that \( \text{Spec}_A \) implements \( S_{NL} \).

To show that \( S_{NL} \) implements \( S_L \), we add to \( \text{Spec}_{NL} \) a prophecy variable \( p \) then a stuttering variable \( s \) to obtain a specification \( \text{Spec}_{ps}^{NL} \) such that \( \exists s, p : \text{Spec}_{ps}^{NL} \) is equivalent to \( \text{Spec}_{NL} \). We then show that \( \exists s, p : \text{Spec}_{ps}^{NL} \) implements \( S_L \) by showing that \( \text{Spec}_{ps}^{NL} \) implements \( \text{Spec}_L \) under a suitable refinement mapping \( \text{mem} \leftarrow \overline{\text{mem}}, \text{istate} \leftarrow \overline{\text{istate}}. \)

The two auxiliary variables we add to \( \text{Spec}_{NL} \) have the following functions:

\( p \) A prophecy variable that predicts for each reader \( i \) which element of the sequence of memory values \( rstate[i] \) will be chosen as the output.

\( s \) A stuttering variable that adds:

- A single stuttering step after a \( \text{BeginRd}(i) \) step if \( p[i] \) predicts that the read will return the current value of memory. The refinement mapping will be defined so that stuttering step will be a \( \overline{\text{DoRd}(i)} \).
- Stuttering steps after a \( \text{DoWr}(i) \) step that will implement the \( \overline{\text{DoRd}(j)} \) step of every current read operation that returns the value of \( \text{mem} \) immediately after the \( \text{DoWr}(i) \) step.

Both these variables are added in a single module named \( \text{NewLinearSnapshotPS} \).

6.5.1 Adding the Prophecy Variable

The prophecy variable \( p \) is a prophecy data structure variable as described in Section 4.4. Its domain \( \text{Dom} \) is the set of readers that are currently executing a read, which can be described as the set of readers \( i \) such that \( rstate[i] \) is a nonempty sequence. The value of \( p[i] \) is a positive integer that predicts which element of the list \( rstate[i] \) will be chosen as the output. This value can be arbitrarily large, since arbitrarily many writes can occur during a read operation, so \( \Pi \) is the set \( \text{Nat} \setminus \{0\} \). Module \( \text{NewLinearSnapshotPS} \) therefore begins

\[
\text{EXTENDS } \text{NewLinearSnapshot} \\
\text{Pi} \triangleq \text{Nat} \setminus \{0\} \\
\text{Dom} \triangleq \{ r \in \text{Readers} : rstate[r] \neq \langle \rangle \} \\
\text{INSTANCE Prophecy WITH DomPrime } \leftarrow \text{Dom'}
\]
It is most convenient to define $p$ in terms of a disjunctive representation in which $EndRd(i)$ is decomposed into $\exists j \in 1..Len(rstate[i]) : IEndRd(i,j)$, where $IEndRd$ can be defined by:

$$IEndRd(i,j) \triangleq \land \text{interface}[i] = \text{NotMemVal}$$
$$\land \text{interface}' = [\text{interface EXCEPT } ![i] = rstate[i][j]]$$
$$\land rstate' = [rstate EXCEPT ![i] = \langle \rangle]$$
$$\land \text{UNCHANGED } \langle \text{mem, wstate} \rangle$$

We could make the change in our original specification $NewLinearSnapshot$, but instead we define a new next-state action $Nxt$ that is equivalent to $Next$:

$$Nxt \triangleq \lor \exists i \in \text{Readers} : \lor \text{BeginRd}(i)$$
$$\lor \exists j \in 1..Len(rstate[i]) : IEndRd(i,j)$$
$$\lor \exists i \in \text{Writers} : \lor \exists cmd \in \text{RegVals} : \text{BeginWr}(i, cmd)$$
$$\lor \text{DoWr}(i) \lor \text{EndWr}(i)$$

It’s easy to see that $Nxt$ is equivalent to formula $Next$ of $NewLinearSnapshot$, and TLAPS can easily check this proof.

**THEOREM** $Next = Nxt$

**BY** DEF $Next$, $Nxt$, $EndRd$, $IEndRd$

A prediction is made for reader $i$ when the element $i$ is added to $Dom$, which is done by a $BeginRd(i)$ step. The prediction is used by the $IEndRd(i,j)$ action, allowing it to be performed only if $p[i]$ has predicted that the $j^{th}$ item in the sequence $rstate[i]$ will be output. The definitions of $Pred_A$, $PredDom_A$, and $DomInj_A$ for the subactions $A$ are given along with the beginning of the module in Figure 22. The module next defines the temporal formula $Condition$, which should be implied by $Spec$. There follows the definition of the specification $SpecP$ obtained by adding the prophecy variable $p$ to $Spec$. TLC can check that $Condition$ is implied by $Spec$, which implies that $\exists p : SpecP$ is equivalent to $Spec$. These definitions appear in Figure 23.

### 6.5.2 Adding the Stuttering Variable

The module next adds the stuttering variable $s$ to $SpecS$. We need to add a single stuttering step after a $BeginRdP(i)$ step iff the reader will output the current value of $mem$, which is the case iff the step sets $p[i]$ to 1. We also need to add a stuttering step after $DoWrP(i)$ for every currently reading reader $j$ for which $p[j]$ predicts that the value of $mem$ that the step appends to $rstate[j]$ is the one that the read will output. These steps are added by letting the values of $s.val$ be subsets of readers, ordered by the subset relation, with the decrement operation removing an element from the set chosen with the CHOOSE operator.

The specification $SpecPS$ obtained by adding the stuttering variable $s$ to $SpecP$ is defined in the part of module $NewLinearSnapshotPS$ shown in Figure 24. The two theorems at the beginning are conditions (5.6) for adding the
\[
\begin{align*}
\text{module NewLinearSnapshotPS} \\
\text{extends NewLinearSnapshot}
\end{align*}
\]

\[
\begin{align*}
\Pi & \triangleq \text{Nat} \setminus \{0\} \\
\text{Dom} & \triangleq \{ r \in \text{Readers} : rstate[r] \neq \langle \rangle \} \\
\text{instance Prophecy with } & \text{DomPrime} \leftarrow \text{Dom}'
\end{align*}
\]

\[
\begin{align*}
\text{IEndRd}(i, j) & \triangleq \land \text{interface}[i] = \text{NotMemVal} \\
& \land \text{interface}' = [\text{interface \ except } ![i] = rstate[i][j]] \\
& \land rstate' = [rstate \ except ![i] = \langle \rangle] \\
& \land \text{UNCHANGED } \langle \text{mem, wstate} \rangle
\end{align*}
\]

\[
\begin{align*}
\text{Nxt} & \triangleq \lor \exists i \in \text{Readers} : \lor \text{BeginRd}(i) \\
& \lor \exists j \in 1 \ldots \text{Len}(rstate[i]) : \text{IEndRd}(i, j) \\
& \lor \exists i \in \text{Writers} : \lor \exists \text{cmd} \in \text{RegVals} : \text{BeginWr}(i, \text{cmd}) \\
& \lor \text{DoWr}(i) \lor \text{EndWr}(i)
\end{align*}
\]

\textbf{THEOREM} \text{Next} = \text{Nxt}

\textbf{BY} \text{DEF} \text{Next, Nxt, EndRd, IEndRd}

\[
\begin{align*}
\text{PredBeginRd}(p) & \triangleq \text{TRUE} \\
\text{PredDomBeginRd} & \triangleq \{\} \\
\text{DomInjBeginRd} & \triangleq \text{IdFcn(\text{Dom})}
\end{align*}
\]

\[
\begin{align*}
\text{PredIEndRd}(p, i, j) & \triangleq j = p[i] \\
\text{PredDomIEndRd}(i) & \triangleq \{i\} \\
\text{DomInjIEndRd} & \triangleq \text{IdFcn(\text{Dom}')}
\end{align*}
\]

\[
\begin{align*}
\text{PredBeginWr}(p) & \triangleq \text{TRUE} \\
\text{PredDomBeginWr} & \triangleq \{\} \\
\text{DomInjBeginWr} & \triangleq \text{IdFcn(\text{Dom})}
\end{align*}
\]

\[
\begin{align*}
\text{PredDoWr}(p) & \triangleq \text{TRUE} \\
\text{PredDomDoWr} & \triangleq \{\} \\
\text{DomInjDoWr} & \triangleq \text{IdFcn(\text{Dom})}
\end{align*}
\]

\[
\begin{align*}
\text{PredEndWr}(p) & \triangleq \text{TRUE} \\
\text{PredDomEndWr} & \triangleq \{\} \\
\text{DomInjEndWr} & \triangleq \text{IdFcn(\text{Dom})}
\end{align*}
\]

\text{Figure 22: Module NewLinearSnapshotPS, part 1.}
\textbf{Condition}  \triangleq  
\[ \square[ \forall i \in \text{Readers} : \]
\[ \land \text{ProphCondition}(\text{BeginRd}(i), \text{DomInjBeginRd}, \]
\[ \text{PredDomBeginRd}, \text{PredBeginRd}) \]
\[ \land \forall j \in 1 \ldots \text{Len}(\text{rstate}[i]) : \]
\[ \text{ProphCondition}(\text{IEndRd}(i, j), \text{DomInjIEndRd}, \]
\[ \text{PredDomIEndRd}(i), \]
\[ \text{LAMBDA } p : \text{PredIEndRd}(p, i, j)) \]
\[ \land \forall i \in \text{Writers} : \]
\[ \land \forall \text{cmd} \in \text{RegVals} : \]
\[ \text{ProphCondition}(\text{BeginWr}(i, \text{cmd}), \text{DomInjBeginWr}, \]
\[ \text{PredDomBeginWr, PredBeginWr}) \]
\[ \land \text{ProphCondition}(\text{DoWr}(i), \text{DomInjDoWr, PredDomDoWr, PredDoWr}) \]
\[ \land \text{ProphCondition}(\text{EndWr}(i), \text{DomInjEndWr, PredDomEndWr, PredEndWr}) \]
\[ ]_{\text{vars}} \]

\textbf{VARIABLE} \ p \\
varsP  \triangleq  \langle \text{vars}, p \rangle \\
\text{InitP}  \triangleq  \text{Init} \land (p = \text{EmptyFcn}) \\
\text{BeginRdP}(i)  \triangleq  \text{ProphAction}(\text{BeginRd}(i), p, p', \text{DomInjBeginRd}, \]
\[ \text{PredDomBeginRd, PredBeginRd}) \]
\[ \text{BeginWrP}(i, \text{cmd})  \triangleq  \text{ProphAction}(\text{BeginWr}(i, \text{cmd}), p, p', \text{DomInjBeginWr}, \]
\[ \text{PredDomBeginWr, PredBeginWr}) \]
\[ \text{DoWrP}(i)  \triangleq  \text{ProphAction}(\text{DoWr}(i), p, p', \text{DomInjDoWr}, \]
\[ \text{PredDomDoWr, PredDoWr}) \]
\[ \text{IEndRdP}(i, j)  \triangleq  \text{ProphAction}(\text{IEndRd}(i, j), p, p', \text{DomInjIEndRd}, \]
\[ \text{PredDomIEndRd}(i), \]
\[ \text{LAMBDA } q : \text{PredIEndRd}(q, i, j)) \]
\[ \text{EndWrP}(i)  \triangleq  \text{ProphAction}(\text{EndWr}(i), p, p', \text{DomInjEndWr}, \]
\[ \text{PredDomEndWr, PredEndWr}) \]
\[ \text{NextP}  \triangleq  \lor \exists i \in \text{Readers} : \lor \exists \text{cmd} \in \text{RegVals} : \text{BeginWrP}(i, \text{cmd}) \]
\[ \lor \exists i \in \text{Writers} : \lor \exists \text{cmd} \in \text{RegVals} : \text{BeginRdP}(i) \]
\[ \lor \exists i \in \text{Readers} : \lor \exists j \in 1 \ldots \text{Len}(\text{rstate}[i]) : \text{IEndRdP}(i, j) \]
\[ \lor \exists i \in \text{Writers} : \lor \exists \text{cmd} \in \text{RegVals} : \text{DoWrP}(i) \lor \text{EndWrP}(i) \]
\[ \text{SpecP}  \triangleq  \text{InitP} \land \square[\text{NextP}]_{\text{varsP}} \land \text{Fairness} \]

\textbf{Figure 23: Module NewLinearSnapshotPS, part 2.}
stuttering steps to \( \text{BeginRdP}(i) \) and \( \text{DoWrP}(i) \) steps. They can be checked by temporarily ending the module immediately after those theorems and running TLC on a model having \( \text{SpecP} \) as its specification. Two assume statements have been added to check the constant conditions on the arguments of the \( \text{MayPostStutter} \) operators used to add those stuttering steps.

### 6.5.3 The Refinement Mapping

Let’s again use the abbreviations \( \text{Spec}_L \) for formula \( \text{Spec} \) of \( \text{LinearSnapshot} \) and \( \text{Spec}_{PS} \) for formula \( \text{Spec}_{PS} \) of module \( \text{NewLinearSnapshot} \). We now define the state functions \( \overline{\text{mem}} \) and \( \overline{\text{istate}} \) such that \( \text{Spec}_{PS} \) implements \( \text{Spec}_L \) under the refinement mapping \( \text{mem} \leftarrow \overline{\text{mem}}, \text{istate} \leftarrow \overline{\text{istate}} \). We want writer actions of \( \text{Spec}_L \) to be simulated by the corresponding writer actions of \( \text{Spec}_{PS} \). Hence, we let \( \overline{\text{mem}} \) equal \( \text{mem} \) and we let \( \overline{\text{istate}}[i] \) equal \( \text{wstate}[i] \) for every writer \( i \). The problem is defining \( \overline{\text{istate}}[i] \) for readers \( i \).

In \( \text{Spec}_L \), for any process \( i \) not executing a read or write, \( \text{istate}[i] \) equals \( \text{interface}[i] \). Hence, we can define \( \overline{\text{istate}}[i] \) to equal \( \text{interface}[i] \) for any reader \( i \) not currently reading. We now consider the case when \( i \) is currently reading, which is true iff \( \text{rstate}[i] \neq \langle \rangle \), which implies \( p[i] \) is a positive integer. There are two possibilities:

- \( p[i] = 1 \) In this case, the \( \text{DoRd}(i) \) step of \( \text{Spec}_L \) is simulated by the stuttering step added to \( \text{BeginRd}(i) \). The \( \text{DoRd}(i) \) step changes \( \text{istate}[i] \) from \( \text{NotMemVal} \) to the memory value to be output, so \( \overline{\text{istate}}[i] \) should equal \( \text{rstate}[i][1] \) when \( \text{rstate}[i] \neq \langle \rangle \), except after the \( \text{BeginRd}(i) \) step and before the stuttering step added immediately after it.

- \( p[i] > 1 \) In this case, the \( \text{DoRd}(i) \) step of \( \text{Spec}_L \) is simulated by one of the stuttering steps added to the \( \text{DoWr}(j) \) step for the writer that appends the \( p[i] \)th element to \( \text{rstate}[i] \). We let it be the stuttering step that removes \( i \) from \( s.\text{val} \), so \( \overline{\text{istate}}[i] \) equals \( \text{NotMemVal} \) until \( p[i] \leq \text{Len}(\text{rstate}[i]) \) and it’s not the case that \( i \) is an element of \( s.\text{val} \) while some writer is performing a stuttering step added after its \( \text{DoWr} \) step.

The definition of \( \overline{\text{istate}} \), under the name \( \text{istateBar} \), appears near the end of the module, shown in Figure 25. The theorems at the end of the module can be checked with TLC. In fact, TLC checks that \( \text{Spec}_{PS} \) satisfies property \( \text{LS}!\text{Spec} \) by checking that the safety part of \( \text{Spec}_{PS} \) satisfies both (a) the safety part of \( \text{LS}!\text{Spec} \) and (b) the property that the liveness part of \( \text{Spec}_{PS} \) implies the liveness part of \( \text{LS}!\text{Spec} \). Therefore, having TLC check that \( \text{Spec}_{PS} \) satisfies \( \text{LS}!\text{Spec} \) checks both theorems.
THEOREM \[ \text{SpecP} \Rightarrow \Box[\forall i \in \text{Readers} : \text{BeginRdP}(i) \Rightarrow \]
\[(\text{if } p'[i] = 1 \text{ then } 1 \text{ else } 0) \in \{0, 1\}]_{\text{varsP}} \]

THEOREM \[ \text{SpecP} \Rightarrow \Box[\forall i \in \text{Writers}, \text{cmd} \in \text{RegVals} : \]
\[ \text{DoWrP}(i) \Rightarrow \]
\[ \{j \in \text{Readers} : (rstate[j] \neq \langle \rangle) \]
\[ \land (p[j] = \text{Len}(rstate'[j]))\} \]
\[ \in (\text{SUBSET \ Readers})]_{\text{varsP}} \]

VARIABLE \( s \)
\[ \text{varsPS} \triangleq \langle \text{vars}, p, s \rangle \]

INSTANCE Stuttering WITH vars ← varsP

\[ \text{InitPS} \triangleq \text{InitP} \land (s = \text{top}) \]

\[ \text{BeginRdPS}(i) \triangleq \text{MayPostStutter}(\text{BeginRdP}(i), \text{"BeginRd"}, i, 0, \]
\[ \text{IF } p'[i] = 1 \text{ then } 1 \text{ else } 0, \]
\[ \text{LAMBDA } j : j - 1) \]

ASSUME StutterConstantCondition(\{0, 1\}, 0, LAMBDA j : j - 1)

\[ \text{BeginWrPS}(i, \text{cmd}) \triangleq \text{NoStutter}(\text{BeginWrP}(i, \text{cmd})) \]

\[ \text{DoWrPS}(i) \triangleq \text{MayPostStutter}(\text{DoWrP}(i), \text{"DoWr"}, i, \{\}, \]
\[ \{j \in \text{Readers} : \]
\[ (rstate[j] \neq \langle \rangle) \land (p[j] = \text{Len}(rstate'[j]))\}, \]
\[ \text{LAMBDA } S : S \setminus \{\text{CHOOSE } x \in S : \text{TRUE}\} \]

ASSUME StutterConstantCondition(\text{SUBSET \ Readers}, \{\}, \]
\[ \text{LAMBDA } S : S \setminus \{\text{CHOOSE } x \in S : \text{TRUE}\} \]

\[ \text{IEndRdPS}(i, j) \triangleq \text{NoStutter}(\text{IEndRdP}(i, j)) \]

\[ \text{EndWrPS}(i) \triangleq \text{NoStutter}(\text{EndWrP}(i)) \]

\[ \text{NextPS} \triangleq \forall \exists i \in \text{Readers} : \forall \text{BeginRdPS}(i) \]
\[ \forall \exists j \in 1 \ldots \text{Len}(rstate[i]) : \text{IEndRdPS}(i, j) \]
\[ \forall \exists i \in \text{Writers} : \forall \exists \text{cmd} \in \text{RegVals} : \text{BeginWrPS}(i, \text{cmd}) \]
\[ \forall \text{DoWrPS}(i) \lor \text{EndWrPS}(i) \]

\[ \text{SafeSpecPS} \triangleq \text{InitPS} \land \Box[\text{NextPS}]_{\text{varsPS}} \]

\[ \text{SpecPS} \triangleq \text{SafeSpecPS} \land \text{Fairness} \]

Figure 24: Module NewLinearSnapshotPS, part 3.
\[ \text{istateBar} \triangleq [i \in \text{Readers} \cup \text{Writers} \mapsto \\
\text{IF } i \in \text{Writers} \\
\text{THEN } \text{wstate}[i] \\
\text{ELSE IF } \text{rstate}[i] = \langle \rangle \\
\text{THEN } \text{interface}[i] \\
\text{ELSE IF } p[i] = 1 \\
\text{THEN IF } \land s \neq \text{top} \\
\land s.id = "\text{BeginRd}" \\
\land s.ctxt = i \\
\text{THEN } \text{NotMemVal} \\
\text{ELSE } \text{rstate}[i][1] \\
\text{ELSE IF } \lor p[i] > \text{Len}(\text{rstate}[i]) \\
\lor \land s \neq \text{top} \\
\land s.id = "\text{DoWr}" \\
\land i \in s.val \\
\text{THEN } \text{NotMemVal} \\
\text{ELSE } \text{rstate}[i][p[i]][i] \]

\[ \text{LS} \triangleq \text{INSTANCE } \text{LinearSnapshot with } \text{istate} \leftarrow \text{istateBar} \]

**Theorem** SafeSpecPS ⇒ LS!SafeSpec

**Theorem** SpecPS ⇒ LS!Spec

Figure 25: Module NewLinearSnapshotPS, part 4.

### 6.6 AfekSimplified Implements NewLinearSnapshot

We now finish checking the correctness of the algorithm Spec\(_A\) of AfekSimplified by showing that it implements \(\exists \text{mem}, \text{rstate}, \text{wstate} : \text{SSpec}_{NL}\), where SSpec\(_{NL}\) is the safety specification of NewLinearSnapshot. As we suggested in Section 6.4, finding a refinement mapping to show this requires adding a history variable to Spec\(_A\) that captures the information remembered by SSpec\(_{NL}\) in the variable rstate. This is straightforward. We just add a history variable h such that h[i] is changed by BeginRd(i), DoWr(i), and an ending TryEndRdH(i) action (one executed with rdVal1[i] = rdVal2[i]) the same way rstate[i] is changed by the corresponding BeginRd(i), DoWr(i), and EndRd(i) action of SSpec\(_{NL}\).

The specification Spec\(_H\), obtained by adding the history variable h to specification Spec of module AfekSimplified, is defined in module AfekSimplified as shown in Figure 26. The definition should be easy to understand by comparing the definition of the initial predicate InitH and of the actions in the module to the corresponding definitions in module NewLinearSnapshot. Note that the action definitions use memBar where the corresponding action definitions in NewLinearSnapshot use mem. The module defines memBar to equal the mem-
MODULE AfekSimplifiedH

EXTENDS AfekSimplified, Sequences

VARIABLE h

varsH \triangleq \langle \text{vars}, h \rangle

InitH \triangleq Init \land (h = [i \in \text{Readers} \mapsto \langle \rangle])

memBar \triangleq [i \in \text{Writers} \mapsto \text{imem}[i][1]]

BeginWrH(i, cmd) \triangleq BeginWr(i, cmd) \land (h' = h)

DoWrH(i) \triangleq \land \text{DoWr}(i)
\land h' = [j \in \text{Readers} \mapsto 
\begin{array}{l}
\text{IF } h[j] = \langle \rangle \\
\quad \text{THEN } \langle \rangle \\
\text{ELSE } \text{Append}(h[j], \text{memBar}')
\end{array}]

EndWrH(i) \triangleq \text{EndWr}(i) \land (h' = h)

BeginRdH(i) \triangleq \land \text{BeginRd}(i)
\land h' = [h \text{ EXCEPT } ![i] = \langle \text{memBar} \rangle]

Rd1H(i) \triangleq \text{Rd1}(i) \land (h' = h)

Rd2H(i) \triangleq \text{Rd2}(i) \land (h' = h)

TryEndRdH(i) \triangleq \land \text{TryEndRd}(i)
\land h' = \text{IF } \text{rdVal1}[i] = \text{rdVal2}[i]
\begin{array}{l}
\text{THEN } [h \text{ EXCEPT } ![i] = \langle \rangle] \\
\text{ELSE } h
\end{array}

NextH \triangleq 
\lor \exists i \in \text{Readers} : \text{BeginRdH}(i) \lor \text{Rd1H}(i) \lor \text{Rd2H}(i) \lor \text{TryEndRdH}(i)
\lor \exists i \in \text{Writers} : \lor \exists cmd \in \text{RegVals} : \text{BeginWrH}(i, cmd)
\lor \text{DoWrH}(i) \lor \text{EndWrH}(i)

SpecH \triangleq \text{InitH} \land \Box[\text{NextH}]_{\text{varsH}}

Figure 26: Beginning of module AfekSimplifiedH.
\[ wstateBar \triangleq [i \in \text{Writers} \mapsto \]
\[
\begin{cases}
  \text{if } (interface[i] = \text{NotRegVal}) \vee (wrNum[i] = imem[i][2]) \\
  \text{THEN } \text{NotRegVal} \\
  \text{ELSE } interface[i]
\end{cases}
\]

\[ NLS \triangleq \text{instance NewLinearSnapshot} \]
\[
\text{with } \text{mem} \leftarrow \text{memBar}, \text{rstate} \leftarrow h, \text{wstate} \leftarrow wstateBar
\]

**THEOREM** \( \text{SpecH} \Rightarrow NLS!SafeSpec \)

Figure 27: End of module \( \text{AfekSimplifiedH} \).

The memory value obtained from \( imem \) in the obvious way, by letting \( \text{memBar}[i] \) equal the first element of \( \text{imem}[i] \). The expression \( \text{memBar} \) is, of course, the value substituted for \( \text{mem} \) by the refinement mapping.

The rest of the refinement mapping is defined at the end of the module, shown in Figure 27. It substitutes \( h \) for \( \text{rstate} \). The expression \( wstateBar \) is substituted for \( \text{wstate} \). To understand it, remember that in \( \text{SSpec}_{\text{NL}} \), the value of \( \text{wstate}[i] \) for a writer \( i \) is \( \text{NotRegVal} \) except between a \( \text{BeginWr}(i, \text{cmd}) \) action and a \( \text{DoWr}(i) \) action, when it equals \( \text{interface}[i] \) (which equals \( \text{cmd} \)).

The theorem was checked by TLC in about 10 hours on a circa 2012 laptop, using a model with: two readers, two writers, and two register values; symmetry of readers and writers (register values aren’t symmetric because \( \text{IntRegVal} \) equals one of them); a state constraint limiting each writer to at most three writes; and two worker threads.
References


[7] Leslie Lamport. TLA—temporal logic of actions. A web page, a link to which can be found at URL [http://lamport.org](http://lamport.org). The page can also be found by searching the Web for the 21-letter string formed by concatenating uid and lamporttlahomepage.
